JAMES A. PETERS

Biostatistical Programs in BASIC Language for Time-Shared Computers: Coordinated with the Book "Quantitative Zoology"

SMITHSONIAN CONTRIBUTIONS TO ZOOLOGY NUMBER 69

SERIAL PUBLICATIONS OF THE SMITHSONIAN INSTITUTION

The emphasis upon publications as a means of diffusing knowledge was expressed by the first Secretary of the Smithsonian Institution. In his formal plan for the Institution, Joseph Henry articulated a program that included the following statement: "It is proposed to publish a series of reports, giving an account of the new discoveries in science, and of the changes made from year to year in all branches of knowledge not strictly professional." This keynote of basic research has been adhered to over the years in the issuance of thousands of titles in serial publications under the Smithsonian imprint, commencing with *Smithsonian Contributions to Knowledge* in 1848 and continuing with the following active series:

Smithsonian Annals of Flight
Smithsonian Contributions to Anthropology
Smithsonian Contributions to Astrophysics
Smithsonian Contributions to Botany
Smithsonian Contributions to the Earth Sciences
Smithsonian Contributions to Paleobiology
Smithsonian Contributions to Zoology
Smithsonian Studies in History and Technology

In these series, the Institution publishes original articles and monographs dealing with the research and collections of its several museums and offices and of professional colleagues at other institutions of learning. These papers report newly acquired facts, synoptic interpretations of data, or original theory in specialized fields. Each publication is distributed by mailing lists to libraries, laboratories, institutes, and interested specialists throughout the world. Individual copies may be obtained from the Smithsonian Institution Press as long as stocks are available.

S. Dillon Ripley
Secretary
Smithsonian Institution

SMITHSONIAN CONTRIBUTIONS TO
ZOOLOGY

NUMBER 69

James A. Peters

Biostatistical Programs in BASIC Language for Time-Shared Computers:

Coordinated with the Book "Quantitative Zoology"

SMITHSONIAN INSTITUTION PRESS
CITY OF WASHINGTON
1971

ABSTRACT

Peters, James A. Biostatistical Programs in "BASIC" Language for Time-Shared Computers: Coordinated with the Book "Quantitative Zoology." *Smithsonian Contributions to Zoology,* 69: 1–00. 1971.—This series of computer programs, written in the language BASIC, is designed to facilitate use of the time-shared computer by individuals with little or no background in computers or programming. The programs are coordinated with the book "Quantitative Zoology," by G. G. Simpson, A. Roe, and R. Lewontin, which permits the user of this publication to find the proper statistical procedures for his needs in that book and to carry out the statistical procedure indicated. The programs are identified by using the page numbers from Simpson et al.

Official publication date is handstamped in a limited number of initial copies and is recorded in the Institution's annual report, Smithsonian Year.

UNITED STATES GOVERNMENT PRINTING OFFICE
WASHINGTON : 1971

For sale by the Superintendent of Documents, U.S. Government Printing Office
Washington, D.C. 20402 - Price 50 cents (paper cover)

TO ALL SCIENTISTS ASSOCIATED WITH THE

MUSEUM OF NATURAL HISTORY

You will find enclosed a copy of a recent publication in the Smithsonian Contributions to Zoology entitled "Biostatistical Programs in BASIC Language for Time-Shared Computers: Coordinated with the Book "Quantitative Zoology." We hope to put a copy in the hands of every scientist in the Museum of Natural History, whether he is an employee of the Smithsonian or not. If you know of anyone who did not receive it please notify James A. Peters (ext. 6171) and one will be sent.

A seminar designed to demonstrate forall interested staff the techniques used to run any of the programs in the enclosed paper will be offered early in September, 1971, when the staff has re-assembled after the summer. A teletype machine will be brought to the seminar room and the entire sequence of steps required to run a program will be followed, so that anyone attending the seminar should then be able to take advantage of the time-shared computer to run his personal data at any time he wishes.

For those already familiar with the teletype, a complete set of tapes of the programs is available. In addition, the Dialcom Company, which currently serves as our time-share contractor, has asked permission to store the entire series in their "library," which will make the programs available to all users whether within the Smithsonian or not. When this has been completed the use of the programs will be even simpler than before. Anyone not familiar with the operation of the time-shared computer who wishes to use it prior to the seminar in September should contact J. A. Peters, as well.

When the seminar is scheduled, a series of additional programs currently available for use by Smithsonian scientists will also be demonstrated. These will include taxonomic identifications from stored matrices, printing of taxonomic diagnoses from the same matrices, automated loan records, stored files and programs to manipulate synonymies and phylogenies, computerized mailing lists, mechanical plotting of graphs, and perhaps an exhibit-oriented question answering service, as well as anything new worked out during the summer.

An announcement of the time and place of the seminar will be made late in the summer.

James A. Peters

Biostatistical Programs in BASIC Language for Time-Shared Computers:

Coordinated with the Book "Quantitative Zoology"

Introduction

This collection of computer programs is designed to make computer use easy for the individual who has no familiarity or knowledge of their internal workings and who does not particularly care to learn about them. For the computer specialist or the biologist who has learned to do his own programming, these programs are of no interest or value at all, and he might even question the value of their existence. But the absolute neophyte can, and I hope will, use this publication as a painless, easy introduction to the computer world. After he has used these programs for awhile, he will find it easy to progress a little further.

The availability of time-shared computers, which are contacted through a familiar typewriter keyboard, now makes most kinds of data manipulation both instantaneous and painless. The investigator who needs to do any kind of work with numbers does himself a serious disservice if he fails to take advantage of computers. But the first hurdle continues to be the highest because it involves the decision to start, and the act of doing so. It involves getting rid of the idea that computer use is complicated and that one cannot do the entire job by himself. It involves learning that the problem need not be large, difficult, and composed of amounts of data that could not be handled in any way other than by computer. Data for computer work can consist of only a few measurements or counts on a dozen specimens, with no more answers wanted than an average, or range, and perhaps the standard deviation. And, finally, it involves spending only fifteen minutes in learning how to use the time-shared computer.

It is my contention that the average biologist does not have to know any more about how to write a program to use the computer than he needs to know about internal gear mechanisms to run a microtome. The set of programs presented here are designed to facilitate such usage. If the reader has ever consulted the book by G. G. Simpson, A. Roe, and R. Lewontin entitled "Quantitative Zoology" (published in 1939, revised in 1960), the use of these programs should be simple. The code designation for the programs are taken directly from the pagination of this book, so that the program entitled "SIP74" will do all the calculations for which a formula is shown on SI(mpson) P(age) 74. If the user of this publication is not familiar with that particular book on biostatistics, he still can use the index to find a program that will run the type of computation he wishes to make. The basic concept behind the use of this book as the locus for organization is that the user will be able to follow

James A. Peters, Department of Vertebrate Zoology, National Museum of Natural History, Smithsonian Institution, Washington, D.C. 20560.

the same procedure as if he were going to calculate with a pencil or with a desk calculator in deciding what formula he wishes to calculate, after which he can substitute the time-shared computer as the device for doing the calculations. In every case the entire program is printed here, including data from an example in "Quantitative Zoology" and also including the result of a run by the computer using those data, which can be directly compared with the book to be sure the program has run correctly. This test is followed by substitution of the user's data (always in line 900 and on), as directed by the program itself, and by a run on that data for his answers.

I visualize these programs as being used in three ways. The most effective way would be for the user to store the entire series in the memory of the computer ("on disc") so that he could simply identify from the book what statistical values he wanted to calculate and then call for the program by using the page number. The first few lines (the "REMS," or remarks) of each program are devoted to instructions on its use, and the user may list these to see what procedures he should follow. His data are then fed in from a tape, and the program is run. If usage by a group is fairly high this would be the best method to use, but if usage is cyclic or erratic the second method probably would be better.

The second way to use these programs would be to have paper tapes of the entire series stored beside the teletype. The user selects the program he needs for the statistical values wanted, feeds it into the computer, and uses it for as long as he has need for it, after which it is removed from the disc storage until the next use. This saves the costs of disc storage but increases the amount of contact time.

The third way to use the programs is directly from the listings presented here. I have tried to provide enough information to permit use even if Simpson's book were not available and the user had to put the program in directly from this booklet. Care must be taken in this case to copy exactly, because slight errors can result in great delays in trying to find them or in totally erroneous answers. Such users should always run the program on the data included here, to see if it is running correctly.

Program lines 14 through 19 have been left vacant throughout these programs so the user can put identification data in them. Anything inserted in these lines with a "PRINT" statement will be put at the head of a sheet of calculated results and will provide a permanent record of the data source. As an example, one might type (on a single line) the following: 15 PRINT "FEMALE PAPILIO, MONTGOMERY COUNTY, WING LENGTH." Everything within the quotation marks would then be printed out with the results.

No attempt is made here to teach the preliminary steps for making contact with the time-shared computer. The techniques and commands change from one contractor to the next, and they generally are taught by the contractor at the beginning of service. Any previous user of a system can show the beginner how to make the first contact.

It will be noted that occasionally there are pages (in Simpson et al.) on which formulae are printed that do not appear here as a numbered program. This is a consequence of the fact that it often is simpler to program a series of operations or formulae as a single unit. The following material is organized to permit the user to find out from the centered headings (all of which refer to pages in Simpson et al.) where material from any page not directly programmed can be found.

SIP1–SIP64

The information contained in these pages, while in many cases amenable to handling, organization, and printing by the computer, generally is such that it is not worth the time or the expense involved in computer use. The techniques described here are used throughout the remainder of the text, and, where appropriate, programs using them have been written. In such cases, the programs often can be used to perform the simple operations required in this section, if the user wishes.

SIP65

The equation presented on this page is included in program SIP74.

SIP66–SIP67

Program SIP74 provides for grouped data as well as for individual values. It will call for an answer from the user as to whether his data is grouped or not. If classes are used as in example 26 on page 67, the class midpoint is to be used as the "X" value in SIP74.

SIP68–SIP73

Program SIP74 will calculate the mean of means, as discussed on these pages. Method 1 sums a series of means and divides the sum by the number of means. If this method is desired, the values of the individual means are entered in line 900, and the ungrouped data routine is chosen when the computer asks for a selection. Method 2 involves multiplying each mean by the size of its sample and then dividing the sum of those values by the sum of the sample sizes. In this method, the user first enters the mean of a sample ("X" value), then the size of the sample ("Y" value), and continue to the end of the samples. The grouped data routine is chosen when the computer offers a selection. If needed by the user, the Median, page 70, can be found in SIP143. The Mode is left for the individual user to tabulate since a computer program scarcely facilitates this operation.

SIP74

```
001 REM  SIP74 CALCULATES VARIOUS MEASURES OF CENTRAL TENDENCY. DATA
002 REM  BEGINS IN LINE 900. FOR INDIVIDUAL VALUES PUT IN ANY SEQUENCE,
003 REM  FOR GROUPED DATA PUT IN FIRST VARIABLE FOLLOWED BY ITS FREQUEN-
004 REM  CY, ETC.  END DATA SERIES WITH THE VALUE 5.0E25.  DATA IN
005 REM  PROGRAM TAKEN FROM P. 66.
006 REM
060 LET B=0
070 LET C=0
080 LET D=0
100 LET J=1
110 LET G=0
200 DIM X(500)
205 PRINT
210 PRINT "TYPE 2 FOR GROUPED DATA, TYPE 1 FOR UNGROUPED DATA"
215 INPUT Q
217 IF Q=2 THEN 400
220 READ X
240 IF X=5.0E25 THEN 500
260 LET C=C+1
290 LET J=J*X
300 LET B=B+(1/X)
320 LET D=D+(X*X)
340 LET G=G+X
360 GO TO 220
400 READ X,Y
410 IF X=5.0E25 THEN 500
420 LET C=C+Y
430 LET G=G+(X*Y)
440 LET B=B+(1/X*Y)
450 LET J=J*(X*Y)
460 LET D=D+(X*X*Y)
470 GO TO 400
500 LET E=1/((1/C)*B)
```

```
700 LET F=SQR(D/C)
760 LET H=G/C
770 LET T=J↑(1/C)
775 PRINT
780 PRINT
790 PRINT "             NUMBER=";C
795 PRINT "ARITHMETIC MEAN=";H
800 PRINT "  HARMONIC MEAN=";E
810 PRINT " QUADRATIC MEAN=";F
830 PRINT " GEOMETRIC MEAN=";T
900 DATA 3.0, 2.8, 3.4, 3.2, 3.0, 2.9, 2.6, 3.3, 3.1, 2.9
901 DATA 2.9, 3.0, 2.8, 2.9, 2.7, 2.9, 3.1, 2.8, 3.0, 3.1, 3.0
903 DATA 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP74

```
TYPE 2 FOR GROUPED DATA, TYPE 1 FOR UNGROUPED DATA
!1


          NUMBER=    21
ARITHMETIC MEAN=     2.97143
  HARMONIC MEAN=     2.96
 QUADRATIC MEAN=     2.97721
 GEOMETRIC MEAN=     2.96569
```

SIP78–SIP95

All of the material contained in Chapter 6 is incorporated in and calculated by SIP87, which also produces results of the equations on pages 166–167 (the standard errors of the median, the first or third quartiles, the mean deviation, the coefficient of variation, and the standard deviation). All of these are based on the same set of data. The early formulae in this chapter are devoted to hand calculation of the equations, and scientists need not spend the amount of time necessary for this when it is so simple today to run the data through a computer. No program is provided for the data manipulation shown on page 95.

SIP87

```
001 REM   SIP87 CALCULATES THE MEASURES OF DISPERSION FOR A SERIES OF
002 REM   VALUES.  DATA BEGIN IN LINE 900. FOR INDIVIDUAL VALUES, PUT
003 REM   IN AS YOU HAVE THEM. FOR GROUPED DATA PUT IN THE FIRST
004 REM   VARIABLE, THEN ITS FREQUENCY, FOLLOWED BY SECOND VARIABLE,
005 REM   ITS FREQUENCY, ETC.  DATA STORED IN PROGRAM FROM P. 86.
006 REM   PROGRAM WILL ALSO GIVE STANDARD ERRORS OF THE VALUES, IF YOU
007 REM   ADD SIP166 TO IT.  DATA STRING IS TERMINATED WITH VALUE
008 REM   5.0E25 IF DATA IS UNGROUPED, SAME VALUE PLUS A ZERO IF
009 REM   DATA IS GROUPED.
010 REM
050 LET A=0
060 LET C=0
070 LET D=0
```

```
080 DIM X(500)
085 PRINT "FOR GROUPED DATA TYPE 2, FOR INDIVIDUAL VALUES TYPE 1"
086 INPUT Q
087 IF Q=2 THEN 205
090 READ X
100 IF X=5.0E25 THEN 220
120 LET A=A+1
150 LET C=C+(X*X)
180 LET D=D+X
200 GO TO 90
205 READ X,Y
206 IF X=5.0E25 THEN 220
208 LET A=A+Y
210 LET C=C+((X*X)*Y)
212 LET D=D+(X*Y)
214 GO TO 205
220 LET F=(C-((D*D)/A))/(A-1)
260 LET G=SQR(F)
265 LET V=(100*G)/(D/A)
270 PRINT
280 PRINT "        NUMBER OF VARIATES=";A
300 PRINT "           ARITHMETIC MEAN=";D/A
310 PRINT "                  VARIANCE =";F
320 PRINT "        STANDARD DEVIATION=";G
360 PRINT "COEFFICIENT OF VARIATION="; V
363 PRINT"  STANDARD ERROR OF MEAN="; G/SQR(A)
900 DATA 52, 1, 54, 3, 56, 3, 57, 8, 58, 7, 59, 11, 60, 11, 61, 10
901 DATA 62, 6, 63, 14, 64, 6, 65, 3, 66, 1, 67, 1, 68, 1, 5.0E25
902 DATA 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP87

```
FOR GROUPED DATA TYPE 2, FOR INDIVIDUAL VALUES TYPE 1
!2

      NUMBER OF VARIATES=    86
         ARITHMETIC MEAN=    60.4302
                VARIANCE=     9.35368
      STANDARD DEVIATION=     3.05838
COEFFICIENT OF VARIATION=     5.06101
  STANDARD ERROR OF MEAN=     .329793
```

SIP90

The calculation of the coefficient of variation has been incorporated in SIP87, since it is based on the same set of data and usually is desired at the same time as the other standard calculations.

SIP113

```
001 REM  SIP113 CALCULATES ESTIMATED SIZE OF A POPULATION BASED ON
002 REM  RECAPTURE OF PREVIOUSLY MARKED INDIVIDUALS.  DATA IS IN-
003 REM  SERTED IN LINE 900 AS FOLLOWS: FIRST VALUE IS THE NUMBER OF
004 REM  INDIVIDUALS FIRST MARKED ("N"), SECOND VALUE IS THE TOTAL
005 REM  NUMBER OF INDIVIDUALS IN RECAPTURE SAMPLE ("R"), THIRD
006 REM  IS NUMBER OF PREVIOUSLY MARKED INDIVIDUALS IN RECAPTURE
007 REM  SAMPLE ("M").  DATA STRING IS ENDED WITH VALUE 5.0E25.  DATA
008 REM  IN PROGRAM IS FROM EXAMPLE 39, P. 114.
050 LET A=0
060 PRINT
070 PRINT
090 PRINT "SAMPLE", "ESTIMATED POPULATION SIZE"
100 READ X, Z, Y
120 IF X=5.0E25 THEN 1000
140 LET A=A+1
160 LET B=(X*Z)/Y
170 PRINT
180 PRINT A, B
200 GØ TØ 100
900 DATA 110, 271, 38, 40, 62, 6, 131, 80, 11, 300, 430, 10, 5.0E25
999 DATA 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP113

```
SAMPLE          ESTIMATED POPULATION SIZE

 1                 784.474

 2                 413.333

 3                 952.728

 4                12900
```

SIP129

```
001 REM  SIP129 CALCULATES THE PROBABILITIES IN A POISSON DISTRIBUTION.
002 REM  DATA IS ASSUMED TØ BE GROUPED, AND MUST BE SO TO RUN IN SIP129.
003 REM  DATA GOES IN LINE 900, WITH VARIABLE FIRST, FOLLOWED BY ITS
004 REM  FREQUENCY, THROUGH ENTIRE SERIES, AND ENDED WITH 5.0E25.
006 REM  STORED DATA FROM EXAMPLE 45, P. 131.
007 REM
050 LET A=0
```

```
060 LET B=0
070 LET C=0
080 LET D=0
090 LET E=0
100 LET F=0
110 LET G=0
120 LET H=1
130 LET I=1
140 LET J=0
163 PRINT
180 DIM X(500), Y(500)
190 READ X, Y
220 IF X=5.0E25 THEN 303
240 LET B=B+(X*Y)
260 LET A=A+((X*X)*Y)
280 LET C=C+Y
300 GØ TØ 190
303 PRINT
315 PRINT
320 LET D=B/C
330 LET Q=-D
335 LET E=2.71828
340 LET E=E↑Q
350 PRINT "     MEAN=";D
355 PRINT "VARIANCE="; (A-(C*(D*D)))/(C-1)
357 PRINT
360 RESTØRE
365 PRINT "FOR AN","   THE POISSON"
370 PRINT "X VALUE","   PROBABILITY"
375 PRINT "  OF:","         IS:"
380 READ X, Y
390 IF X=5.0E25 THEN 1000
400 LET F=X
420 IF F=0 THEN 550
440 LET G=(F-1)
450 IF G=0 THEN 570
460 LET H=H*G
480 LET I=F*H
500 LET J=(E*(D↑F))/I
515 PRINT
520 PRINT F,J
540 GØ TØ 380
550 PRINT F,E
555 PRINT
560 GØ TØ 380
570 PRINT F,D*E
580 GØ TØ 380
900 DATA 0,16,1,9,2,3,3,1,4,1,5,0,5.0E25
999 DATA 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP129

```
     MEAN=        .733334
 VARIANCE=       1.02989

 FOR AN            THE POISSON
 X VALUE           PROBABILITY
    OF:                 IS:
     0                .480306

     1                .352224

     2                .129149

     3             .315697E-01

     4             .578778F-02

     5             .848874E-03
```

SIP143

```
001 REM   SIP143 RUNS THE COEFFICIENT OF SKEWNESS.  PUT GROUPED DATA
002 REM   IN LINE 900, FIRST THE VARIABLE, SECOND THE FREQUENCY.
003 REM   TERMINATE DATA WITH VALUE 5.0E25.  DATA IN PROGRAM TAKEN
004 REM   FROM EXAMPLE 43, P. 144.
005 REM
050 LET B=0
052 LET D=0
056 LET H=0
058 LET J=1
061 LET T=0
070 DIM X(500), Y(500)
080 READ X, Y
090 IF X=5.0E25 THEN 130
105 LET D=D+Y
110 LET T=T+((X*X)*Y)
120 LET E=E+(X*Y)
125 GØ TØ 80
130 LET F=E/D
133 PRINT
135 PRINT "NUMBER=";D
140 PRINT "    MEAN=";F
210 LET S=(T-((E*E)/D))/(D-1)
220 LET U=SQR(S)
225 PRINT "STD.DEV.=";U
227 PRINT
230 LET I=D/2
235 RESTØRE
237 READ X1, Y1, X2, Y2
```

```
238 LET B=X2-X1
250 RESTØRE
260 READ X, Y
280 LET H=H+Y
290 IF H>=I THEN 350
300 GØ TØ 260
350 PRINT "VALUE OF MEDIAN CLASS IS:";X
370 LET H=H-Y
390 LET H=H+1
400 IF H>=I THEN 450
410 LET J=J+1
420 GØ TØ 390
450 PRINT "SERIAL NUMBER OF MEDIAN VALUE IN ITS CLASS=";J
460 LET L=X-(B/2)
480 LET L=L+((J-.5)*B)/Y
500 PRINT "PRECISE ESTIMATE OF MEDIAN="; L
520 LET N=(3*(F-L))/U
540 PRINT "COEFFICIENT OF SKEWNESS=";N
900 DATA 5.5, 3, 5.6, 12, 5.7, 43, 5.8, 80, 5.9, 131, 6.0, 236
901 DATA 6.1, 185, 6.2, 142, 6.3, 99, 6.4, 37, 6.5, 15, 6.6, 12
902 DATA 6.7, 3, 6.8, 2, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP143

```
NUMBER=  1000
   MEAN=      6.06149
STD.DEV.=      .201288

VALUE OF MEDIAN CLASS IS:     6
SERIAL NUMBER OF MEDIAN VALUE IN ITS CLASS:   231
PRECISE ESTIMATE OF MEDIAN:     6.04767
COEFFICIENT OF SKEWNESS=      .206068
```

SIP146

```
001 REM  SIP146 RUNS THE COEFFICIENT OF KURTOSIS.  PUT GROUPED DATA
002 REM  IN LINE 900, FIRST THE VARIABLE, AND SECOND THE FREQUENCY.
003 REM  TERMINATE DATA WITH THE VALUE 5.0E25.  DATA IN PROGRAM TAKEN
004 REM  FROM EXAMPLE 51, P. 147.
005 REM
050 LET C=0
052 LET H=0
057 LET I=0
058 LET L=0
100 READ X, Y
140 IF X=5.0E25 THEN 280
150 LET C=C+Y
```

```
160 LET I=I+(X*X*Y)
170 LET H=H+(X*Y)
260 GØ TØ 100
280 LET D=H/C
300 LET E=(I-((H*H)/C))/(C-1)
320 LET F=SQR(E)
400 RESTØRE
420 READ X, Y
440 IF X=5.0E25 THEN 480
460 LET G=X-D
470 LET L=L+(G↑4*Y)
475 GØ TØ 420
480 LET M=(L/(F↑4*C))-3
540 PRINT
550 PRINT "                    NUMBER=";C
560 PRINT "           ARITHMETIC MEAN=";D
565 PRINT "                  VARIANCE=";E
570 PRINT "        STANDARD DEVIATION=";F
580 PRINT "COEFFICIENT OF KURTOSIS=";M
900 DATA 52, 1, 54, 3, 56, 3, 57, 8, 58, 7, 59, 11, 60, 11, 61, 10
901 DATA 62, 6, 63, 14, 64, 6, 65, 3, 66, 1, 67, 1, 68, 1, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP146

```
                 NUMBER=    86
        ARITHMETIC MEAN=    60.4302
               VARIANCE=     9.35368
     STANDARD DEVIATION=     3.05838
COEFFICIENT OF KURTOSIS=    -.106298
```

SIP161

```
020 REM  SIP161 CALCULATES THE CONFIDENCE LIMITS OF THE VARIANCE.
021 REM  YOU MUST SUPPLY THE CHI SQUARE VALUES FOR THE LIMITS YOU
022 REM  WISH TO ESTABLISH.  THIS PROGRAM IS WRITTEN TO BE ADDED TO
023 REM  SIP87, WHICH CALCULATES THE VARIANCE.  IT CANNOT BE RUN
024 REM  INDEPENDENTLY OF THAT PROGRAM NOR AT THE SAME TIME AS SIP163.
025 REM
382 PRINT "WHAT IS CHI SQUARE OF LOWER CONFIDENCE LIMIT?"
384 INPUT R1
386 PRINT "WHAT IS CHI SQUARE OF UPPER CONFIDENCE LIMIT?"
388 INPUT R2
389 PRINT
390 PRINT "CONFIDENCE LIMITS FOR VARIANCE ARE:"
392 PRINT ((A-1)*F)/R2 " TO" ((A-1)*F)/R1
395 PRINT
400 PRINT"CONFIDENCE LIMITS FOR STANDARD DEVIATION ARE:"
410 PRINT SQR(((A-1)*F)/R2)"  TO" SQR(((A-1)*F)/R1)
```

Sample Run of SIP161

```
FOR GROUPED DATA TYPE 2, FOR INDIVIDUAL VALUES TYPE 1
!2

        NUMBER OF VARIATES=      86
           ARITHMETIC MEAN=      60.4302
                  VARIANCE=       9.35368
        STANDARD DEVIATION=       3.05838
  COEFFICIENT OF VARIATION=       5.06101
    STANDARD ERROR OF MEAN=        .329793
WHAT IS CHI SQUARE OF LOWER CONFIDENCE LIMIT?
!61.36
WHAT IS CHI SQUARE OF UPPER CONFIDENCE LIMIT?
!112.476

CONFIDENCE LIMITS FOR VARIANCE ARE:
     7.06873  TO    12.9573

CONFIDENCE LIMITS FOR STANDARD DEVIATION ARE:
     2.65871    TO      3.59963
```

SIP163

```
030 REM  SIP163 WILL CALCULATE THE CHI SQUARE VALUES FOR THE CONFIDENCE
031 REM  LIMITS YOU SET, AND ADD THEM TO THE RESULTS OF SIP87.  YOU
032 REM  MUST INPUT THE AMOUNT OF THE AREA OF THE NORMAL CURVE YOU
033 REM  WISH TO INCLUDE.  THIS PROGRAM CANNOT BE RUN INDEPENDENTLY
034 REM  OF SIP87, BUT MUST BE MERGED WITH IT.  IT CANNOT BE RUN AT
035 REM  THE SAME TIME AS SIP161.
036 REM
382 PRINT "WHAT IS VALUE OF NORMAL VARIATE YOU HAVE SELECTED?"
384 INPUT N
385 PRINT "THE CHI SQUARE VALUES ARE:"
386 LET R1=((-N+SQR(2*(A-1)))↑2)/2
387 LET R2=((N+SQR(2*(A-1)))↑2)/2
388 PRINT R1 "AND"  R2
389 PRINT
390 PRINT "CONFIDENCE LIMITS FOR VARIANCE ARE:"
392 PRINT ((A-1)*F)/R2"  TO" ((A-1)*F)/R1
395 PRINT
400 PRINT"CONFIDENCE LIMITS FOR STANDARD DEVIATION ARE:"
410 PRINT SQR(((A-1)*F)/R2)"  TO" SQR(((A-1)*F)/R1)
```

Sample Run of SIP163

```
FOR GROUPED DATA TYPE 2, FOR INDIVIDUAL VALUES TYPE 1
!2
```

```
         NUMBER OF VARIATES=     86
            ARITHMETIC MEAN=     60.4302
                   VARIANCE=      9.35368
         STANDARD DEVIATION=      3.05838
   COEFFICIENT OF VARIATION=      5.06101
     STANDARD ERROR OF MEAN=       .329793
WHAT IS VALUE OF NORMAL VARIATE YOU HAVE SELECTED?
!1.96
THE CHI SQUARE VALUES ARE:
    61.3655  AND   112.476

CONFIDENCE LIMITS FOR VARIANCE ARE:
     7.06872   TO    12.9562

CONFIDENCE LIMITS FOR STANDARD DEVIATION ARE:
     2.65871   TO     3.59947
```

SIP166

```
008 REM  SIP166 WILL CALCULATE THE STANDARD ERRORS OF VARIOUS
009 REM  MEASURES OF DISPERSAL PLUS THE VALUES OF ANY GIVEN CONFI-
010 REM  DENCE LIMITS TO THOSE STANDARD ERRORS.  THIS PROGRAM MUST BE
011 REM  ADDED TO SIP87 TO RUN.  IT CANNOT BE RUN INDEPENDENTLY.
012 REM  IF YOU WANT CALCULATED CONFIDENCE LIMITS YOU WILL BE ASKED
013 REM  TO SUPPLY EITHER THE 'T' VALUE OR THE AMOUNT OF THE NORMAL
020 REM  CURVE WITHIN LIMITS YOU ESTABLISH, TO PERMIT CALCULATION
021 REM  OF THE LIMITS.
500 PRINT
503 PRINT "STANDARD ERRORS:"
505 PRINT "     MEDIAN=";(1.2533*G)/SQR(A)
510 PRINT "     FIRST ØR THIRD QUARTILE=";(1.3636*G)/SQR(A)
520 PRINT "     MEAN DEVIATION=";(.6028*G)/SQR(A)
540 PRINT "     COEFFICIENT OF VARIATION="; V/Z
560 PRINT "     STANDARD DEVIATION=";G/Z
565 PRINT "CONFIDENCE LIMITS WANTED? 2 FOR YES, 1 FOR NO."
570 INPUT K
575 IF K=1 THEN 1000
580 PRINT "DEGREES OF FREEDOM ARE:";A-1
585 PRINT"WHAT IS T VALUE OR AMT OF NORMAL CURVE FOR CONF. LIMITS?"
590 INPUT T
600 PRINT
610 PRINT "CONFIDENCE LIMITS ARE:"
620 PRINT "MEAN: PLUS-MINUS"; (T*(G/SQR(A)))
640 PRINT "MEDIAN: PLUS-MINUS"; (T*1.2533*G)/SQR(A)
645 PRINT "QUARTILES: PLUS-MINUS"; (T*1.3636*G)/SQR(A)
650 PRINT "M. D.: PLUS-MINUS"; (T*.6028*G)/SQR (A)
660 PRINT "C.V.: PLUS-MINUS"; (T*(V/Z))
700 GØ TØ 1000
```

Sample Run of SIP166

```
FOR GROUPED DATA TYPE 2, FOR INDIVIDUAL VALUES TYPE 1
!2
      NUMBER OF VARIATES=    86
         ARITHMETIC MEAN=    60.4302
                VARIANCE=     9.35368
      STANDARD DEVIATION=     3.05838
COEFFICIENT OF VARIATION=     5.06101

STANDARD ERRORS:
    MEDIAN=        .41333
    FIRST ØR THIRD QUARTILE=       .449706
    MEAN DEVIATION=       .198799
    COEFFICIENT OF VARIATION=       .385898
    STANDARD DEVIATION=       .233199
CONFIDENCE LIMITS WANTED? 2 FOR YES, 1 FOR NO.
! 2
DEGREES ØF FREEDOM ARE:     85
WHAT IS T VALUE OR AMT OF NORMAL CURVE FOR CONF. LIMITS?
!1.96
CONFIDENCE LIMITS ARE:
MEAN: PLUS-MINUS        .646395
MEDIAN: PLUS-MINUS        .810127
QUARTILES: PLUS-MINUS        .881425
M. D.: PLUS-MINUS        .389647
C.V.: PLUS-MINUS        .756361
```

SIP176

```
001 REM  SIP176 IS SET UP TO CALCULATE BOTH STUDENT'S 'T' AND
002 REM  THE 'F' VALUE FOR TWO SAMPLES TO BE COMPARED.  IF YOU WANT
003 REM  ONLY STUDENT'S 'T', TYPE THE FOLLOWING LINE OF INSTRUCTION:
004 REM    755 GØ TØ 1000
005 REM  IF YOU WANT ONLY THE VALUE OF 'F', TYPE THE FOLLOWING:
006 REM    600 GØ TØ 760
007 REM  DATA CAN BE GROUPED OR UNGROUPED. IT IS PUT IN LINE 900.
008 REM  IF UNGROUPED PUT THOSE OF FIRST SAMPLE IN, TERMINATED WITH
009 REM  VALUE 2.0E25, THEN SECOND SAMPLE, ENDING WITH 5.0E25.
010 REM  IF GROUPED, DO SAME, BUT PUT VALUE FIRST FOLLOWED BY ITS
011 REM  FREQUENCY, AND DOUBLE TERMINATION VALUES IN BOTH CASES.
050 LET A=0
051 LET C=0
052 LET D=0
053 LET J=0
054 LET H=0
055 LET L=0
```

```
060 PRINT
070 PRINT "TYPE 1 FOR UNGROUPED DATA; 2 FOR GROUPED DATA"
080 INPUT K
090 IF K=2 THEN 160
100 READ T
110 IF T=2.0E25 THEN 220
130 LET A=A+1
140 LET D=D+T
150 LET C=C+(T*T)
155 GØ TØ 100
160 READ X,Y
165 IF X=1.0E25 THEN 220
170 LET A=A+Y
180 LET D=D+(X*Y)
190 LET C=C+(X*X*Y)
200 GØ TØ 160
220 LET F=(C-((D*D)/A))/(A-1)
260 LET G=SQR(F)
280 LET S1=D/A
285 PRINT
286 PRINT
290 PRINT "                    FIRST SAMPLE"
300 PRINT "                        MEAN=";S1
310 PRINT "                    VARIANCE=";F
320 PRINT "      STANDARD DEVIATION=";G
330 PRINT "STANDARD ERROR OF MEAN=";G/SQR(A)
340 IF K=2 THEN 440
350 READ T
360 IF T=5.0E25 THEN 480
380 LET L=L+1
400 LET H=H+T
420 LET J=J+(T*T)
430 GØ TØ 350
440 READ X,Y
450 IF X=5.0E25 THEN 480
460 LET L=L+Y
465 LET H=H+(X*Y)
470 LET J=J+(X*X*Y)
475 GØ TØ 440
480 LET V=(J-((H*H)/L))/(L-1)
520 LET M=SQR(V)
540 LET S2=H/L
545 LET W=F/V
550 PRINT
555 PRINT"                     SECØND SAMPLE"
560 PRINT "                        MEAN=";S2
570 PRINT "                    VARIANCE=";V
580 PRINT "      STANDARD DEVIATION=";M
590 PRINT "STANDARD ERROR OF MEAN=";M/SQR(L)
640 LET N=(S1-S2)*SQR((A*L)/(A+L))
660 LET Ø=((A-1)*F)+((L-1)*V)
720 LET T=N/(SQR(Ø/(A+L-2)))
730 PRINT
```

```
740 PRINT "STUDENTS T=";T
750 PRINT "DEGREES OF FREEDOM=";A+L-2
760 PRINT
770 PRINT "VALUE OF F RATIO FOR SAMPLES IS:";W
775 PRINT "DEGREES OF FREEDOM IN NUMERATOR=";A-1
776 PRINT "DEGREES OF FREEDOM IN DENOMINATOR=";L-1
900 DATA 50.5, 50.0, 50.1, 48.5, 49.5, 49.2, 49.7, 2.0E25
901 DATA 47.2, 48.0, 47.9, 48.5, 47.4, 47.8, 50.0, 50.2, 47.4
902 DATA 48.8, 48.3, 47.6, 49.1, 50.2, 49.7, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP176

```
TYPE 1 FOR UNGROUPED DATA; 2 FOR GROUPED DATA
!1


                        FIRST SAMPLE
                           MEAN=      49.6428
                       VARIANCE=        .434245
             STANDARD DEVIATION=        .658973
         STANDARD ERROR OF MEAN=        .249068

                        SECOND SAMPLE
                           MEAN=      48.54
                       VARIANCE=        1.14118
             STANDARD DEVIATION=        1.06826
         STANDARD ERROR OF MEAN=        .275824

STUDENTS T=     2.49963
DEGREES OF FREEDOM=     20

VALUE OF F RATIO FOR SAMPLES IS:      .380522
DEGREES OF FREEDOM IN NUMERATOR=     6
DEGREES OF FREEDOM IN DENOMINATOR=     14
```

SIP181

```
001 REM  SIP181 RUNS THE STUDENT'S 'T' TEST FOR PAIRED SAMPLES. THE
002 REM  TWO VALUES IN EACH PAIR ARE ENTERED IN LINE 900, FOLLOWED
003 REM  BY THE NEXT PAIR. SERIES IS ENDED WITH VALUE 5.0E25 REPEATED
004 REM  ONCE.  DATA IN PROGRAM FROM EXAMPLE 58, P. 181.
005 REM
050 LET A=0
060 LET C=0
070 LET D=0
100 READ X, Y
120 IF X=5.0E25 THEN 240
140 LET A=A+1
```

```
160 LET B=X-Y
180 LET C=C+B
200 LET D=D+(B*B)
220 GØ TØ 100
240 LET E=(D-((C*C)/A))/(A-1)
260 LET T=(C/A)/SQR(E/A)
320 PRINT "VALUE OF STUDENTS T IS:";T
330 PRINT "DEGREES OF FREEDOM=";A-1
900 DATA 9.8, 10.2, 10.5, 10.7, 10.5, 10.7 , 10.8, 10.8, 11.0, 11.0
901 DATA 11.1, 11.4, 11.1, 12.1, 11.3, 12.6, 11.4, 12.8, 11.4, 10.8
902 DATA 11.4, 12.6, 11.9, 12.3, 12.2, 12.4, 12.2, 12.0, 12.3, 13.7
903 DATA 12.3, 13.0, 12.4, 13.2, 12.4, 12.4, 12.5, 13.8
904 DATA 12.7, 13.5, 12.8, 13.3, 13.0, 12.7, 13.1, 13.1, 13.2, 13.6
906 DATA 13.4, 12.6, 13.5, 13.5, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP181

```
VALUE OF STUDENTS T IS=     -3.41602
DEGREES OF FREEDOM=    25
```

SIP183

```
001 REM   SIP183 TESTS SINGLE INDIVIDUALS WITH THE KNOWN
002 REM   VALUES OF A PREVIOUS SAMPLE.  DATA GOES IN LINE 900 AND
003 REM   SO ON, USING THE FOLLOWING SEQUENCE OF KNOWN VALUES:
004 REM   FIRST NUMBER=MEAN OF KNOWN SAMPLE
005 REM   SECOND NUMBER=STANDARD DEVIATION OF KNOWN SAMPLE
006 REM   THIRD NUMBER=VALUE FOR INDIVIDUAL SPECIMEN
007 REM   FOURTH NUMBER=NUMBER OF SPECIMENS IN KNOWN SAMPLE
008 REM   THIS IS THEN FOLLOWED BY SAME SEQUENCE FOR SECOND
009 REM   CHARACTERISTIC TO BE TESTED AGAINST KNOWN SAMPLE.
010 REM   SEQUENCE IS ENDED WITH VALUE 5.0E25.  DATA STORED IN PROGRAM
011 REM   FROM EXAMPLE 59, P. 183.
012 REM
050 LET D=0
055 PRINT
060 PRINT "VARIATE", "SAMPLE MEAN","SAMPLE S. D.","    STUDENTS T"
080 DIM X(100), Y(100), Z(100), W(100)
100 READ X, Y, Z, W
120 IF X=5.0E25 THEN 1000
140 LET D=D+1
160 LET A=X-Z
180 LET B=SQR(W/(W+1))
220 LET T=(A*B)/Y
240 PRINT
250 PRINT D,X,Y,T
260 GØ TØ 100
```

```
900 DATA 13.6, .7, 14.3, 15, 9.6, .5, 9.4, 16, 16.4, .8, 16.7
901 DATA 15, 11.9, .7, 10.9, 15, 5.0E25
999 DATA 5.0E25, 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP183

VARIATE	SAMPLE MEAN	SAMPLE S. D.	STUDENTS T
1	13.6	.7	-.968245
2	9.6	.5	.388058
3	16.4	.8	-.363091
4	11.9	.7	1.38321

SIP185

The calculation of the ratio between sample variances ("F") is included in SIP176, since it is based on the same set of data. Instructions will be found in that program to permit calculation of both Student's *t* and "F," or either one without the other, as the user wishes.

SIP187

```
001 REM  SIP187 PERMITS CALCULATION OF CHI SQUARE AND GIVES YOU A
002 REM  CHOICE OF INCLUDING YATE'S CORRECTION (P. 190) FOR ANY
003 REM  NUMBER OF SAMPLES, ALL OF WHICH WILL BE PAIRED WITH ALL OTHERS.
004 REM  DATA IS PUT IN LINE 900, WITH THE TWO VALUES FOR THE FIRST
005 REM  LOCALITY ENTERED, THEN THE TWO FOR THE SECOND LOCALITY, AND
006 REM  SO ON.  SERIES IS TERMINATED WITH THE VALUE 5.0E25.
007 REM  DATA IN PROGRAM FROM EXAMPLE ON P. 188.
060 LET Q=1
070 LET F=1
090 PRINT
100 PRINT "WHAT IS NUMBER OF SAMPLES?"
110 INPUT A
120 PRINT"IF YATES CORRECTION IS WANTED, TYPE 2, IF NOT, TYPE 1"
130 INPUT B
135 PRINT
140 FØR K=1 TØ (A+1)
150 READ X(K), Y(K)
160 IF X(K)=5.0E25 THEN 600
170 IF K<Q THEN 470
180 IF K>Q THEN 220
190 LET R=X(K)
200 LET S=Y(K)
205 LET F=Q
```

```
210 GØ TØ 470
220 LET F=F+1
240 LET N=S+R+X(K)+Y(K)
250 LET P=(R*Y(K))-(S*X(K))
260 LET E=(P*P*N)/((R+S)*(X(K)+Y(K))*(R+X(K))*(S+Y(K)))
300 PRINT"FOR SAMPLE";Q;"AND SAMPLE";F;" CHI SQUARE IS:";E
380 IF P<0 THEN 480
400 LET M=P-(N/2)
410 LET J=(M*M*N)/((R+S)*(X(K)+Y(K))*(R+X(K))*(S+Y(K)))
415 IF B=1 THEN 470
420 PRINT"  CHI SQUARE WITH YATES CONTINUITY CORRECTION IS:";J
430 PRINT
470 NEXT K
480 LET M=P+(N/2)
500 GØ TØ 410
600 IF Q=A THEN 1000
610 LET Q=Q+1
620 RESTØRE
630 GØ TØ 140
900 DATA 22, 24, 23, 33, 50, 15, 5.0E25
999 DATA 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP187

```
WHAT IS NUMBER OF SAMPLES?
!3
IF YATES CORRECTION IS WANTED, TYPE 2, IF NOT, TYPE 1
! 2

FOR SAMPLE      1 AND SAMPLE      2  CHI SQUARE IS:       .467375
  CHI SQUARE WITH YATES CONTINUITY CORRECTION IS:        .233549

FOR SAMPLE      1 AND SAMPLE      3  CHI SQUARE IS:     10.0068
  CHI SQUARE WITH YATES CONTINUITY CORRECTION IS:       8.77075

FOR SAMPLE      2 AND SAMPLE      3  CHI SQUARE IS:     16.1563
  CHI SQUARE WITH YATES CONTINUITY CORRECTION IS:     14.693
```

SIP189–SIP190

Yates's correction for continuity is included in SIP187. The user is asked if he wishes to make a correction because of sample size during the run of that program; a "yes" answer will run Yates's correction on the data.

SIP191–SIP212

The material in these pages is covered for the most part by SIP176. Since the amount of data does not affect the computer operation unless it is extremely large, the simpler formula for Student's *t* on SIP194 is not needed.

SIP223

```
001 REM  SIP223 RUNS A REGRESSION ON ANY PAIR OF VALUES FOR ANY NUMBER
002 REM  OF INDIVIDUALS.  DATA GOES IN LINE 900, FIRST X VALUE, THEN
003 REM  Y, FOLLOWED BY NEXT X AND NEXT Y.  FINISH SERIES WITH VALUE
004 REM  5.0E25 REPEATED ONCE.  SIP223 ALSO PRINTS THE CORRELATION
005 REM  COEFFICIENT (P. 238), THE 'Z' VALUE (P. 244), AND THE Z
006 REM  DIVIDED BY S SUB Z (P. 245).  DATA IN PROGRAM FROM EXAMPLE
007 REM  65, P. 222.
008 REM
050 LET E=0
060 LET C=0
070 LET G=0
080 LET I=0
100 READ X,Y
120 IF X=5.0E25 THEN 300
140 LET A=A+1
180 LET C=C+X
220 LET E=E+Y
240 LET G=G+(X*Y)
250 LET I=I+(X*X)
260 LET L=L+(Y*Y)
290 GØ TØ 100
300 LET B=(G-((C*E)/A))/(I-((C*C)/A))
320 LET D=(G-((C*E)/A))/(L-((E*E)/A))
340 LET F=(E/A)-(B*(C/A))
360 LET H=(C/A)-(D*(E/A))
380 LET S1=(I-((C*C)/A))/(A-1)
390 LET S2=(L-((E*E)/A))/(A-1)
400 LET T=((A-1)/(A-2))*(S2-(B*B*S1))
420 LET U=((A-1)/(A-2))*(S1-(D*D*S2))
425 LET R=SQR(B*D)
430 PRINT
431 PRINT "NUMBER OF VARIATES=";A
434 PRINT "   MEAN OF X=";C/A,       "   MEAN OF Y=";E/A
435 PRINT "   VARIANCE OF X=";S1,    "   VARIANCE OF Y=";S2
436 PRINT
440 PRINT "VALUES FOR REGRESSION OF Y ON X ARE:"
441 PRINT
450 PRINT        "   Y-INTERCEPT=";F
460 PRINT              "   SLOPE=";B
470 PRINT         "   REGRESSION EQUATION IS: Y=";B;" X  +";F
480 PRINT "   VARIANCE OF ESTIMATE OF Y ON X=";T
481 PRINT
490 PRINT "VALUES FOR REGRESSION OF X ON Y ARE:"
500 PRINT
510 PRINT              "   X-INTERCEPT=";H
520 PRINT              "   SLOPE=";D
530 PRINT "   REGRESSION EQUATION IS:X=";D; "  Y  + ";H
540 PRINT        "   VARIANCE OF ESTIMATE OF X ON Y=";U
545 PRINT
```

```
550 PRINT "THE CORRELATION COEFFICIENT=";R
555 LET Z=(LØG(1+R)-LØG(1-R))/2
560 PRINT "THE Z VALUE=";(LØG(1+R)-LØG(1-R))/2
570 PRINT "Z/S OF Z=";Z/(1/SQR(A-3))
900 DATA 37, 284, 49, 375, 50, 353, 51, 366, 53, 418, 54, 408
901 DATA 68, 510, 86, 627, 93, 683, 106, 820, 130, 1056, 137, 986
902 DATA 142, 1086, 142, 1086, 146, 1078, 149, 1122, 155, 1254
903 DATA 156, 1202, 187, 1387, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP223

```
NUMBER OF VARIATES=     19
   MEAN OF X=     104.789       MEAN OF Y=    794.79
   VARIANCE OF X=  2261.62    VARIANCE OF Y=   .134401E 06

VALUES FOR REGRESSION OF Y ON X ARE:

   Y-INTERCEPT=     -9.8822
   SLOPE=      7.67893
   REGRESSION EQUATION IS: Y=      7.67893 X   +     -9.8822
   VARIANCE OF ESTIMATE OF Y ON X=   1103.39

VALUES FOR REGRESSION OF X ON Y ARE:

   X-INTERCEPT=      2.08945
   SLOPE=       .129217
   REGRESSION EQUATION IS:X=       .129217  Y   +       2.08945
   VARIANCE OF ESTIMATE OF X ON Y=     18.5677

THE CORRELATION COEFFICIENT=       .996116
THE Z VALUE=      3.12099
Z/S OF Z=     12.484
```

SIP225

The variance of the estimate is included in SIP223, and it will be included in the output of any run of that program (see above).

SIP226

```
001 REM  SIP226 CALCULATES THE CONFIDENCE INTERVALS FOR THE INTER-
002 REM  CEPT (ALPHA) AND THE SLOPE (BETA) IN A REGRESSION.  THIS
003 REM  PROGRAM MUST BE ADDED TO SIP223, AND CANNOT BE RUN INDEPEN-
004 REM  DENTLY, SINCE IT USES THE DATA AND RESULTS OF THAT
005 REM PROGRAM.  YOU WILL BE ASKED TO SUPPLY THE 'T' VALUE FOR THE
006 REM  CONFIDENCE LIMITS YOU SELECT.  TEST RUN RESULTS SHOWN IN
007 REM  EXAMPLE 67, P. 228.
```

```
600 PRINT "DEGREES OF FREEDOM ARE:"; A-2
610 PRINT
620 PRINT "WHAT IS T VALUE FOR CONFIDENCE LIMITS SELECTED?"
630 INPUT T1
633 LET E1=(1/A)+(((C/A)↑2)/((A-1)*S1))
635 LET P1=SQR(E1)
640 PRINT "CONFIDENCE LIMITS ARE:"
650 PRINT "   FOR Y ON X:"
660 PRINT " ",B" PLUS-MINUS" (T1*SQR(T))/(SQR(S1)*SQR(A-1))
670 PRINT " ",F" PLUS-MINUS" T1*SQR(T)*P1
680 PRINT "   FOR X ON Y:"
690 PRINT " ",D " PLUS-MINUS" (T1*SQR(U))/(SQR(S2)*SQR(A-1))
700 PRINT " ",H " PLUS-MINUS" (T1*SQR(U))/P1
```

Sample Run of SIP226

```
NUMBER OF VARIATES=    19
   MEAN OF X=   104.789        MEAN OF Y=    794.79
  VARIANCE OF X=  2261.62    VARIANCE OF Y=   .134401E 06

VALUES FOR REGRESSION OF Y ON X ARE:

   Y-INTERCEPT=     -9.8822
   SLOPE=      7.67893
   REGRESSION EQUATION IS: Y=      7.67893 X   +      -9.8822
  VARIANCE OF ESTIMATE OF Y ON X=  1103.39

VALUES FOR REGRESSION OF X ON Y ARE:

   X-INTERCEPT=      2.08945
   SLOPE=       .129217
   REGRESSION EQUATION IS:X=       .129217  Y  +       2.08945
   VARIANCE OF ESTIMATE OF X ON Y=     18.5677

THE CORRELATION COEFFICIENT=       .996116
THE Z VALUE=      3.12099
Z/S OF Z=    12.484
DEGREES OF FREEDOM ARE:     17

WHAT IS T VALUE FOR CONFIDENCE LIMITS SELECTED?
! 2.11
CONFIDENCE LIMITS ARE:
   FOR Y ON X:
                     7.67893  PLUS-MINUS      .347378
                   -9.8822   PLUS-MINUS    39.7947
  FOR X ON Y:
                     .129217 PLUS-MINUS  .584553E-02
                    2.08945  PLUS-MINUS    16.0134
```

SIP229

```
001 REM  SIP229 USES DATA TAKEN FROM TWO REGRESSIONS RUN IN SIP223.
002 REM  DATA GOES IN LINE 900, AS FOLLOWS:
003 REM     FIRST--NUMBER OF VARIATES
004 REM     SECOND--VARIANCE OF X (OR OF Y)
005 REM     THIRD--SLOPE
006 REM     FOURTH--VARIANCE OF ESTIMATE
007 REM  THIS IS THEN FOLLOWED BY SAME VALUES FOR SECOND REGRESSION.
008 REM  DATA STORED IN THIS PROGRAM IS TAKEN FROM EXAMPLE 68, P. 230.
009 REM
020 PRINT
100 READ N1, S1, B1, T1, N2, S2, B2, T2
120 LET A=((N1-1)*S1)+((N2-1)*S2)
140 LET D=(N1-1)*(N2-1)*S2*S1
160 LET C=(((N1-2)*T1)+((N2-2)*T2))/(N1+N2-4)
180 LET T=((B1-B2)*SQR(D/A))/SQR(C)
260 PRINT "STUDENTS T=";T
270 PRINT
280 PRINT "DEGREES OF FREEDOM=";N1+N2-4
900 DATA 24, 98917, .153, 79.45, 19, 134401, .129, 32.36
1000 END
```

Sample Run of SIP229

```
STUDENTS T=      3.38546

DEGREES OF FREEDOM=    39
```

SIP232

```
001 REM  SIP232 PERMITS CALCULATIONS OF PREDICTED VALUES FOR THE
002 REM  DEPENDENT VARIABLE FROM THE REGRESSION EQUATION.  IT IS
003 REM  ASSUMED THE NECESSARY REGRESSION HAS BEEN CALCULATED, AND
004 REM  THE VALUES NEEDED ARE KNOWN.  PUT THE VALUE OF THE INTER-
005 REM  CEPT (A-SUB-Y) AND OF THE SLOPE (B-SUB-Y) OF Y ON X IN THE
006 REM  DATA LINE 900, FOLLOWED BY ANY VALUES OF X FOR WHICH A
007 REM  PREDICTED VALUE OF Y IS DESIRED.  PROCEDURE SHOULD BE
008 REM  REVERSED FOR PREDICTION OF X VALUES FROM KNOWN Y VALUES.
009 REM  END DATA STRING WITH VALUE 5.0E25.  DATA FROM EXAMPLE 65,
010 REM  P. 222.
011 PRINT
050 PRINT "FOR THE KNOWN",     "    PREDICTED"
051 PRINT "    VALUE,",          "    VALUE IS:"
100 READ A, B
150 LET C=(-A/B)+(1/B)
200 READ Y
225 IF Y=5.0E25 THEN 1000
250 LET D=C*Y
```

```
300 PRINT
320 PRINT Y,"",D
400 GØ TØ 200
900 DATA -11.4, 7.68, 2, 3, 4, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP232

FOR THE KNOWN VALUE,	PREDICTED VALUE IS:
2	3.22917
3	4.84375
4	6.45833

SIP233

```
001 REM  SIP233 RUNS BARTLETT'S BEST FIT LINE, AND THE CONFIDENCE
002 REM  LIMITS OF THE SLOPE (BETA) AND THE INTERCEPT (ALPHA, P.
003 REM  236).  IT ALSO CALCULATES THE VARIANCES AS SHOWN ON P.
004 REM  236, BUT DOES NOT PRINT THEM.  DATA GOES IN LINE 900, AS IN
005 REM  EXAMPLE 69, P. 234.   TERMINATE FIRST GROUP OF DATA WITH THE
006 REM  REPEATED VALUE 2.0E25, SECOND WITH REPEATED 3.0E25, THIRD
007 REM  WITH 5.0E25. VALUE OF STUDENT'S 'T' IS LOOKED UP IN TABLE ON
008 REM  P. 422, USING DEGREES OF FREEDOM GIVEN BY PROGRAM AND CONFI-
009 REM  DENCE LIMITS YOU SELECT.
050 LET A=0
055 LET B=0
060 LET C=0
062 LET D=0
064 LET E=0
066 LET F=0
068 LET G=0
070 LET H=0
072 LET I=0
100 READ X,Y
110 IF X=2.0E25 THEN 220
140 LET L=L+1
160 LET M=M+X
180 LET C=C+Y
200 GØ TØ 100
220 LET J=M/L
240 LET K=C/L
300 READ X,Y
310 IF X=3.0E25 THEN 420
340 LET D=D+1
```

```
360 LET E=E+X
380 LET F=F+Y
400 GØ TØ 300
420 LET U=E/D
440 LET Q=F/D
500 READ X,Y
520 IF X=5.0E25 THEN 620
540 LET G=G+1
560 LET H=H+X
580 LET I=I+Y
600 GØ TØ 500
620 LET N=H/G
640 LET Ø=I/G
720 LET P=L+G+D
760 LET B=(Ø-K)/(N-J)
780 LET A=((C+F+I)/P)-((M+E+H)/P*B)
790 PRINT
800 PRINT "THE EQUATION OF BARTLETTS BEST FIT LINE IS:"
810 PRINT "    Y=";A;"    +";B;" X"
813 PRINT
815 PRINT "DEGREES OF FREEDOM =";P-3
817 PRINT
820 PRINT "IF YOU WANT CONFIDENCE INTERVALS, TYPE 2, IF NOT TYPE 1"
825 INPUT Z
830 IF Z=1 THEN 1000
835 PRINT "GIVE VALUE OF STUDENTS T FOR LIMITS YOU WISH"
840 INPUT E
841 RESTØRE
843 PRINT
845 READ X,Y
850 IF X=2.0E25 THEN 860
855 LET S1=S1+((X-J)*(X-J))
856 LET S2=S2+((X-J)*(Y-K))
857 LET S3=S3+((Y-K)*(Y-K))
859 GØ TØ 845
860 READ X,Y
861 IF X=3.0E25 THEN 870
862 LET S1=S1+((X-U)*(X-U))
863 LET S2=S2+((X-U)*(Y-Q))
864 LET S3=S3+((Y-Q)*(Y-Q))
865 GØ TØ 860
870 READ X,Y
871 IF X=5.0E25 THEN 880
872 LET S1=S1+((X-N)*(X-N))
873 LET S2=S2+((X-N)*(Y-Ø))
874 LET S3=S3+((Y-Ø)*(Y-Ø))
875 GØ TØ 870
876 PRINT
880 LET V=1/(P-3)
881 LET S1=S1*V
882 LET S2=S2*V
883 LET S3=S3*V
885 LET Y=(E*E*S1)-(((N-J)↑2)*(G/2))
```

```
887 LET R=(-2*E*E*S2)+(2*((N-J)↑2)*B*(G/2))
889 LET C=((N-J)↑2*(G/2)*B*B)-(E*E*S3)
890 PRINT "SMALL A="; Y/Y
891 PRINT "SMALL B=";R/Y
892 PRINT "SMALL C=";-C/Y
893 PRINT
895 LET T=SQR(((R/Y)↑2)-(4*(-C/Y)))/2
896 PRINT "CONFIDENCE LIMITS OF SLOPE:";B;"  PLUS-MINUS";T
897 LET  W=(E/SQR(P))*SQR((S3-(2*B*S2)+(B*B*S1)))
898 PRINT "CONFIDENCE LIMITS OF INTERCEPT:";A;"  PLUS-MINUS"; W
900 DATA 37, 284, 49, 375, 50, 353, 51, 366, 53, 418, 54, 408
901 DATA 2.0E25, 2.0E25, 68, 510, 86, 627, 93, 683, 106, 820
902 DATA 130, 1056, 137, 986, 142, 1086, 3.0E25, 3.0E25, 142, 1086
903 DATA 146, 1078, 149, 1122, 155, 1254, 156, 1202, 187, 1387
904 DATA 5.0E25, 5.0E25
999 DATA 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP233

```
THE EQUATION OF BARTLETTS BEST FIT LINE IS:
   Y=   -10.3402    +      7.6833 X

DEGREES OF FREEDOM =     16

IF YOU WANT CONFIDENCE INTERVALS, TYPE 2, IF NOT TYPE 1
!2
GIVE VALUE OF STUDENTS T FOR LIMITS YOU WISH
!2.12

SMALL A=      1
SMALL B=    -15.3681
SMALL C=     58.8827

CONFIDENCE LIMITS OF SLOPE:       7.6833  PLUS-MINUS        .402287
CONFIDENCE LIMITS OF INTERCEPT:   -10.3402  PLUS-MINUS     16.6322
```

SIP234–SIP237

Most of the material in these pages is calculated in SIP233, although not all of it is printed out. If the user wishes, a series of "PRINT" statements following his final data line will give the following values from the program:

$\overline{X}_1 = J$	$N = P$	$\overline{Y}_1 = K$	$s_x^2 = S1$
$\overline{X}_2 = U$	$K = G$	$\overline{Y}_2 = Q$	$s_{xy} = S2$
$\overline{X}_3 = N$	$A = A$	$\overline{Y}_3 = Ø$	$s_y^2 = S3$
$\overline{X} = (M+E+H)/P$	$B = B$	$\overline{Y} = (C+F+I)/P$	

A "PRINT" statement for any one of these would look like this:

925 PRINT "X BAR SUB 1 (MEAN OF FIRST GROUP) =", J

The Student's *t* called for in this section (and in SIP233) must always be looked up in the table on page 422, using the known degrees of freedom (printed out by SIP233) and the confidence limits the user wishes to establish. This value *cannot* be calculated by the program; it must be typed as input by the user when it is called for by program (see above).

SIP238

```
001 REM  SIP238 GIVES CONFIDENCE LIMITS FOR THE MEAN VALUE OF ANY
002 REM  PREDICTED Y FOR ANY GIVEN VALUE OF X, AND ALSO THE LIMITS
003 REM  FOR ACTUAL VALUE OF Y FOR SAME VALUE OF X, IN A REGRESSION.
004 REM  DATA FOR THIS PROGRAM COMES FROM THE PRINTOUT OF SIP223,
005 REM  WHICH INCLUDES ALL NECESSARY INFORMATION. THE STUDENT'S
006 REM  'T' CALLED FOR IS LOOKED UP IN THE TABLE ON P. 422, USING
007 REM  KNOWN DEGREES OF FREEDOM AND DESIRED CONFIDENCE LIMITS. THE
008 REM  DATA TO TEST THIS PROGRAM SHOULD BE INSERTED FROM EXAMPLE
009 REM  72, P. 239.
010 REM
090 PRINT
100 PRINT "ENTER THE FOLLOWING VALUES IN THIS SEQUENCE:"
101 PRINT "NUMBER, MEAN, VARIANCE, SLOPE, Y-INTERCEPT, VARIANCE"
102 PRINT "OF ESTIMATE, AND T VALUE"
110 INPUT A, B, C, G, H, D, E
115 PRINT
120 PRINT "ENTER VALUE OF X (IF AT END OF X VALUES, TYPE 5.0E25)"
125 INPUT X
126 IF X=5.0E25 THEN 1000
130 LET F=(1/A)+(((X-B)*(X-B)))/((A-1)*C)
140 LET N=E*D*(SQR(1+F))
150 LET F=E*D*(SQR(F))
180 LET M=H+(G*X)
199 PRINT
200 PRINT "FOR GIVEN X, Y=";M
201 PRINT
210 PRINT "CONFIDENCE LIMITS FOR MEAN Y=";M-F;"  --";M+F
211 PRINT
220 PRINT "CONFIDENCE LIMITS FOR INDIVIDUAL PREDICTED"
240 PRINT "VALUE ØF Y=";M-N;"  --";M+N
250 GØ TØ 115
1000 END
```

Sample Run of SIP238

```
ENTER THE FOLLOWING VALUES IN THIS SEQUENCE:
NUMBER, MEAN, VARIANCE, SLOPE, Y-INTERCEPT, VARIANCE
OF ESTIMATE, AND T VALUE
!19, 104.78, 2261.61, 7.68, -11.4, 32.72, 2.11
```

```
ENTER VALUE OF X (IF AT END OF X VALUES, TYPE 5.0E25)
!1150

FOR GIVEN X, Y=  1140.6

CONFIDENCE LIMITS FOR MEAN Y=  1118.46  --  1162.74

CONFIDENCE LIMITS FOR INDIVIDUAL PREDICTED
VALUE OF Y=  1068.1  --  1213.1

ENTER VALUE OF X (IF AT END OF X VALUES, TYPE 5.0E25)
!5.0E25
```

SIP239–SIP245

The calculation of the correlation coefficient ("r") is included in program SIP223, which utilizes the same set of data needed for the calculation of this value. It will be printed out along with the other results when that program is run. The transformation of "r" into Fisher's "z" value also is included in SIP223, as is the test of the significance of z, as shown in Example 74, page 245.

SIP246

```
001 REM  SIP246 CALCULATES THE SIGNIFICANCE OF THE DIFFERENCE BETWEEN
002 REM  TWO CORRELATION COEFFICIENTS USING THE TRANSFORMED CORREL-
003 REM  ATION COEFFICIENT ('Z').  DATA FOR THIS PROGRAM IS DERIVED
004 REM  FROM TWO SEPARATE REGRESSIONS RUN IN SIP223.  PUT DATA IN
005 REM  LINE 900: FIRST, NUMBER OF VARIATES IN FIRST GROUP, ITS
006 REM  CORRELATION COEFFICIENT ('R'), AND ITS TRANSFORMED CORREL-
007 REM  ATION COEFFICIENT ('Z'), THEN THE SAME THREE VALUES FOR
008 REM  THE SECOND REGRESSION.  DATA IN PROGRAM TAKEN FROM EXAMPLE
009 REM  75, P. 246.
010 REM
100 READ N1, R1, Z1, N2, R2, Z2
120 LET A=(1/(N1-3))+(1/(N2-3))
140 LET A=(Z1-Z2)/SQR(A)
170 PRINT
180 PRINT
200 PRINT "THE APPROXIMATE STANDARDIZED NORMAL DEVIATE=";A
91 DATA 24, .974, 2.16, 19, .988, 2.56
1000 END
```

Sample Run of SIP246

```
THE APPROXIMATE STANDARDIZED NORMAL DEVIATE=    -1.20539
```

SIP248

```
001 REM  SIP248 GIVES PARTIAL CORRELATION FOR ANY NUMBER OF VARIATES.
002 REM  WHEN INPUT IS CALLED FOR, GIVE THE CORRELATION COEFFICIENT
003 REM  (R) FOR THE VARIATES FOR WHICH PARTIAL CORRELATION IS TO BE
004 REM  TESTED, FOLLOWED BY CØRR. COEFF. OF EACH OF THESE VARIABLES
```

```
005 REM  WITH THE VARIABLE WHOSE EFFECT IS TO BE ELIMINATED (AS
006 REM  DISCUSSED ON P. 248).  REPEAT EACH TIME INPUT IS CALLED
007 REM  FOR, CHANGING THE SEQUENCE UNTIL ALL PARTIAL CORRELATIONS
008 REM  HAVE BEEN DETERMINED.  DATA TØ TEST PROGRAM SHOULD BE TAKEN
009 REM  FROM EXAMPLE ON P. 250.
010 REM
080 PRINT
085 PRINT
090 PRINT "IDENTIFICATION NUMBER OF TEST (5.0E25 IF AT END OF TESTS)"
091 INPUT D
092 IF D=5.0E25 THEN 1000
095 PRINT
100 PRINT"ENTER R VALUE FOR COMBINATION OF VARIABLES TO BE TESTED FOR"
101 PRINT"PARTIAL CORRELATION, THEN R VALUE OF EACH WITH VARIABLE TO"
102 PRINT"BE ELIMINATED (TWO VALUES):"
120 INPUT A, B, C
180 LET M=(1-(B*B))*(1-(C*C))
200 LET M=(A-(B*C))/SQR(M)
290 PRINT
300 PRINT "PARTIAL CORRELATION FOR";D;"IS:";M
301 PRINT
320 GØ TØ 90
1000 END
```

Sample Run of SIP248

```
IDENTIFICATION NUMBER OF TEST (5.0E25 IF AT END OF TESTS)
!1

ENTER R VALUE FOR COMBINATION OF VARIABLES TO BE TESTED FOR
PARTIAL CORRELATION, THEN R VALUE OF EACH WITH VARIABLE TO
BE ELIMINATED (TWO VALUES):
! .355, .795, -.046

PARTIAL CORRELATION FOR     1 IS:      .64619

IDENTIFICATION NUMBER OF TGST (5.0E25 IF AT END OF TESTS)
! 2

ENTER R VALUE FOR COMBINATION OF VARIABLES TO BE TESTED FOR
PARTIAL CORRELATION, THEN R VALUE OF EACH WITH VARIABLE TO
BE ELIMINATED (TWO VALUES):
! .795, .355, -.046

PARTIAL CORRELATION FOR     2 IS:      .868777

IDENTIFICATION NUMBER OF TEST (5.0E25 IF AT END OF TESTS)
!3

ENTER R VALUE FOR COMBINATION OF VARIABLES TO BE TESTED FOR
PARTIAL CORRELATION, THEN R VALUE OF EACH WITH VARIABLE TO
BE ELIMINATED (TWO VALUES):
! -.046, .355, .795
```

```
PARTIAL CORRELATION FOR      3 IS:      -.57878

IDENTIFICATION NUMBER OF TEST (5.0E25 IF AT END OF TESTS)
5.0E25
```

SIP254

No program has been prepared for the formula on page 254. While it is feasible, such a program would be long and cumbersome, and to calculate the "C" value it would appear simpler to use the method as shown in the example on page 255.

SIP270

```
001 REM  SIP270 RUNS AN ANALYSIS OF VARIANCE, ONE-FACTOR DESIGN.
002 REM  DATA BEGINS IN LINE 900, WITH EACH SERIES OF INDIVIDUAL
003 REM  VARIABLES ENDING WITH VALUE 2.0E25, INCLUDING FINAL SERIES.
004 REM  AFTER FINAL 2.0E25, PUT IN THE VALUE 5.0E25.  DATA STORED IN
005 REM  PROGRAM IS FROM EXAMPLE 80, P. 272.
006 REM
022 PRINT
023 PRINT
024 LET I=0
025 LET C=0
026 LET D=0
027 LET E=0
028 LET F=0
029 PRINT "  ","      TOTAL"
030 PRINT "BASIC DATA","      VALUE","   NUMBER","    MEAN"
035 LET A=0
036 LET B=0
037 LET N=0
095 LET C=C+1
100 READ X
105 IF X=5.0E25 THEN 300
110 IF X=2.0E25 THEN 160
120 LET A=A+X
130 LET B=B+(X*X)
140 LET N=N+1
150 GØ TØ 100
160 PRINT
170 PRINT "FOR SERIES";C; A, N, A/N
180 LET D=D+A
200 LET E=E+B
220 LET F=F+N
225 LET H=(A*A)/N
230 LET I=I+H
240 GØ TØ 35
300 LET G=(D*D)/F
310 LET K=E-G
320 LET J=I-G
340 LET L=E-I
360 LET M=C-2
```

```
380 LET N=F-(C-1)
390 PRINT
400 PRINT
420 PRINT " ","    SUM OF"," ","    MEAN"
430 PRINT "SOURCE","    SQUARES","   D.F.","    SQUARE"
440 PRINT
450 PRINT "MAIN EFFECT",J,M,J/M
460 PRINT"DEVIATIONS",L,N,L/N
500 PRINT
510 PRINT "F RATIO="; (J/M)/(L/N)
900 DATA 120, 120, 121, 122, 122, 122, 123, 125, 125, 126, 126
901 DATA 2.0E25, 123, 124, 125, 125, 126, 127, 127, 127, 128, 128
902 DATA 129, 129, 2.0E25, 122, 122, 125, 127, 127, 127, 128
903 DATA 129, 2.0E25, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP270

```
                              TOTAL
BASIC DATA                    VALUE      NUMBER         MEAN

FOR SERIES          1         1352        11           122.909

FOR SERIES          2         1518        12           126.5

FOR SERIES          3         1007         8           125.875


                           SUM OF                       MEAN
SOURCE                     SQUARES       D.F.           SQUARE

MAIN EFFECT               81              2             40.5
DEVIATIONS               140.813         28              5.02902

F RATIO=        8.05326
```

SIP276

```
001 REM  SIP276 RUNS AN ANALYSIS OF VARIANCE IN REGRESSION (TABLE, P
002 REM  274) USING MACHINE FORMULAE (P. 276).  TO RUN, PUT
003 REM  IN DATA STARTING WITH LINE 900, WITH VALUE OF X FIRST AND
004 REM  VALUE OF Y SECOND.  TERMINATE EACH SET OF XS AND  YS WITH
005 REM  THE VALUE 2.0E25 REPEATED ONCE, AND TERMINATE THE FINAL SET
006 REM  WITH 2.0E25 TWICE AND 5.0E25 TWICE.  STORED DATA FROM P. 278.
007 REM
015 PRINT
020 LET C=0
021 LET P=0
```

```
023 LET N=0
024 LET L=0
025 LET K=0
026 LET H=0
027 LET J=0
028 LET I=0
029 PRINT" ", " ", "MEAN";"     MEAN";"    TOTAL";"   TOTAL"
030 PRINT" ","     NUMBER";"  OF X";"     OF Y";"    OF X";"    OF Y"
035 LET A=0
036 LET B=0
037 LET F=0
038 LET G=0
039 LET D=0
040 LET E=0
090 LET C=C+1
100 READ X, Y
110 IF X=5.0E25 THEN 400
120 IF X=2.0E25 THEN 260
130 LET F=F+1
140 LET A=A+X
160 LET B=B+Y
180 LET G=G+(X*Y)
200 LET D=D+(X*X)
220 LET E=E+(Y*Y)
230 LET Q1=F*X
235 LET Q2=F*(X↑2)
240 GØ TØ 100
260 PRINT
261 PRINT "FOR GROUP"; C; F; (A/F); (B/F); A; B
263 LET M=M+(E/F)
265 LET N=N+((B*B)/F)
270 LET L=L+E
280 LET K=K+D
290 LET W1=W1+Q1
300 LET H=H+F
310 LET W2=W2+Q2
320 LET I=I+B
340 LET J=J+A
350 LET P=P+G
360 GØ TØ 35
400 LET B=(P-((J*I)/H))/(K-((J*J)/H))
420 LET A=(I/H)-(B*(J/H))
440 PRINT
450 PRINT "THE REGRESSION EQUATION IS: Y=";A;"  +";B;" X"
500 LET G=L-N
520 LET D=(B*B*W2)-(B*B*((W1*W1)/H))
540 LET E=N-((I*I)/H)-D
550 LET T=E/((C-1)-2)
560 LET V=G/(H-(C-1))
598 PRINT
599 PRINT
600 PRINT "SOURCE","  SUM OF","  DEGREES OF","   MEAN"
610 PRINT " ",    "  SQUARES","  FREEDOM","   SQUARE"
```

```
620 PRINT
630 PRINT "IN LEVELS",G, H-(C-1),G/(H-(C-1))
640 PRINT"REGRESSIONS",D, "      1", D
650 PRINT"DEVIATIONS", E, (C-1)-2, E/((C-1)-2)
680 PRINT
685 PRINT "TOTAL", L-((I*I)/H), H-1
695 PRINT
700 PRINT "F RATIO (SIGNIFICANCE OF REGRESSION COEFFICIENT)=";D/T
705 PRINT "F RATIO (TEST FOR RECTILINEARITY)="; T/V
900 DATA 1, 139.7, 1, 127.0, 1, 133.4, 1, 177.8, 2.0E25,2.0E25
901 DATA 2, 139.7, 2, 215.9, 2, 171.5, 2, 152.4, 2, 228.6, 2
902 DATA 190.5, 2, 149.2, 2.0E25, 2.0E25, 3, 203.2, 3, 241.3, 3
903 DATA 209.6, 3, 215.9, 3, 190.5, 2.0E25, 2.0E25, 4, 241.3, 4
904 DATA 247.7, 4, 235.0, 2.0E25, 2.0E25, 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP276

```
                        NUMBER   MEAN     MEAN      TOTAL   TOTAL
                                 OF X     OF Y      OF X    OF Y

FOR GROUP     1           4        1      144.475     4      577.9

FOR GROUP     2           7        2      178.257    14     1247.8

FOR GROUP     3           5        3      212.1      15     1060.5

FOR GROUP     4           3        4      241.333    12      724

THE REGRESSION EQUATION IS: Y=    112.8  +    32.6 X

SOURCE          SUM OF        DEGREES OF      MEAN
                SQUARES       FREEDOM         SQUARE

IN LEVELS      10207.8          15            680.517
REGRESSIONS    19577.1           1          19577.1
DEVIATIONS        25.6563        2             12.8281

TOTAL          29810.5          18

F RATIO (SIGNIFICANCE OF REGRESSION COEFFICIENT)=   1526.11
F RATIO (TEST FOR RECTILINEARITY)=   .188506E-01
```

SIP283

```
001 REM  SIP283 PERMITS CALCULATION OF AN ANALYSIS OF VARIANCE FOR
002 REM  A TWO FACTOR DESIGN, AS IN EXAMPLE 82, P. 284.  PUT DATA IN
003 REM  LINE 900, LISTING VARIABLES FROM EACH CELL AS A CONTINUOUS
004 REM  SERIES, FOLLOWED IMMEDIATELY BY VARIABLES OF SECOND SERIES,
```

```
005 REM  AND SO ON.  NOTE THAT THE PROGRAM WILL CALL FOR INPUT FOR
006 REM  YOU TO SUPPLY.
100 PRINT "WHAT IS NUMBER OF SAMPLES IN EACH CELL?"
110 INPUT A
120 PRINT "WHAT IS NUMBER OF CELLS?"
130 INPUT B
140 FØR J=1 TØ (B*2)
150 FØR I = 1 TØ A
161 READ Z(I,J)
165 LET T=T+Z(I,J)
167 LET R=R+Z(I,J)*Z(I,J)
170 NEXT I
180 NEXT J
184 RESTØRE
190 FØR K=1 TØ 2
200 FØR J=1 TØ B
210 FØR I=1 TØ A
220 READ X
230 LET D=D+X
250 NEXT I
260 LET W(J,K)=D
270 LET V(J,K)=D*D
275 LET E=V(J,K) + E
277 LET D=0
280 NEXT J
290 NEXT K
350 LET P=R-(T*T)/(A*B*2)
355 LET K=1
360 FØR J=1 TØ B
380 LET S=W(J,K)+W(J,K+1)
410 LET Q=Q+(S*S)
420 LET S=0
430 NEXT J
440 LET S=(Q/(A*2))-((T*T)/(A*B*2))
450 FØR K=1 TØ 2
460 FØR J=1 TØ B
470 LET N=N+W(J,K)
480 NEXT J
485 LET N=N*N
490 LET M=M+N
500 LET N=00
510 NEXT K
600 LET N=(M/(B*A))-((T*T)/(A*B*2))
625 LET L=(E/A)-(Q/(A*2))-(M/(B*A))+((T*T)/(A*B*2))
650 LET F=R-(E/A)
651 LET X=B-1
652 LET Y=2*B*(A-1)
653 LET U=S+N+L+F
654 LET T=1
690 PRINT
691 PRINT
700 PRINT "",           "SUM OF",    "DEG. OF",    "MEAN"
701 PRINT "SOURCE:",  "SQUARES",   "FREEDOM",   "SQUARE",   "  F VALUES"
```

```
702 PRINT
703 PRINT  "ROWS",        S,           X,          S/X,     (S/X)/(F/Y)
706 PRINT
710 PRINT "COLUMNS",      N,           T,           N,         N/(F/Y)
711 PRINT
720 PRINT "INTERACTION", L  ,        X,          L/X,     (L/X)/(F/Y)
721 PRINT
730 PRINT "DEVIATIONS",  F,            Y,          F/Y
731 PRINT
740 PRINT@"TOTALS",       U,         X+X+1+Y
900 DATA 7, 19, 18, 9, 1, 15, 29, 114, 24, 37, 49, 64, 124, 63
901 DATA83, 51, 81, 106, 72, 100, 67, 87, 68, 9, 25, 16, 10, 9
902 DATA 28, 14, 35, 22, 18, 45, 29, 27, 20, 26, 38, 44, 127, 52
903 DATA 40, 263, 129, 45, 100, 115
1000 END
```

Sample Run of SIP283

```
WHAT IS NUMBER OF SAMPLES IN EACH CELL?
!6
WHAT IS NUMBER OF CELLS?
!4
```

SOURCE:	SUM OF SQUARES	DEG. OF FREEDOM	MEAN SQUARE	F VALUES
ROWS	39930.9	3	13310.3	9.59612
COLUMNS	8.34375	1	8.34375	.601546E-02
INTERACTION	12066	3	4022	2.89968
DEVIATIONS	55482	40	1387.05	
TOTALS	.107487E 06	47		

SIP287–SIP297

The programs for the calculations on these pages have not been written, primarily because the authors have given no example that would permit the verification of any program prepared. If any user of this manual has actually calculated and verified an analysis of variance for a three-factor design and if he will submit the data and answers to the author, a program can be prepared. The same can be done for the random models, if a user needs them.

SIP300

```
001 REM  SIP300 PERMITS CALCULATION OF A HIERARCHICAL ANALYSIS OF
002 REM  VARIANCE.  PUT DATA IN FOLLOWING SEQUENCE, STARTING IN LINE
003 REM  900: FIRST, ALL SAMPLES FROM ONE LOCALITY, THEN SAMPLES FROM
```

```
004 REM  EACH ADDITIONAL LOCALITY IN SAME SEQUENCE AS FIRST.  DATA
005 REM  IN PROGRAM FROM EXAMPLE 83, ON P. 302.
006 REM
100 PRINT "WHAT IS NUMBER OF LOCALITIES?"
110 INPUT A
120 PRINT "WHAT IS NUMBER OF SUBSAMPLES PER LOCALITY?"
130 INPUT B
140 PRINT "WHAT IS NUMBER OF SPECIMENS IN EACH SUBSAMPLE?"
150 INPUT C
160 FØR J=1 TØ A
170 FØR I=1 TØ B
180 FØR K=1 TØ C
190 READ X
200 LET S=S+X
210 LET W=X*X
220 LET V=V+W
230 LET U=U+X
240 NEXT K
250 LET P=P+(S*S)
260 LET S=0
270 NEXT I
280 LET T=T+U
290 LET R=R+(U*U)
300 LET U=0
310 NEXT J
320 LET S=R/(B*C)-(T*T/(A*B*C))
330 LET M=P/C-(R/(B*C))
340 LET N=V-(P/C)
350 LET L=V-(T*T/(A*B*C))
360 LET F1=A-1
370 LET F2=A*(B-1)
380 LET F3=A*B*(C-1)
390 PRINT  "","SUM ØF","  DEG. OF","     MEAN"
391 PRINT "SØURCE","SQUARES","  FREEDOM","     SQUARE"
392 PRINT
400 PRINT "LOCALITIES",S,F1,S/F1
401 PRINT
402 PRINT "SAMPLES",M,F2,M/F2
405 PRINT
410 PRINT "SPECIMENS",N,F3,N/F3
411 PRINT
420 PRINT "TOTALS",L,F1+F2+F3
421 PRINT
422 PRINT
430 PRINT "F-TESTS:"
440 PRINT "  FOR LOCALITIES, F=";(S/F1)/(M/F2)
441 PRINT "    WITH DEGREES ØF FREEDOM:";F1;"AND";F2
450 PRINT
451 PRINT "  FOR SAMPLES, F=";(M/F2)/(N/F3)
452 PRINT "    WITH DEGREES ØF FREEDØM:";F2;"AND";F3
900 DATA 27, 31, 30, 30, 27
901 DATA 26, 28, 29, 31, 29
902 DATA 28, 31, 31, 28, 33
```

```
904 DATA 29, 25, 28, 27, 30
905 DATA 35, 33, 33, 35, 38
906 DATA 33, 33, 31, 33, 37
907 DATA 32, 36, 33, 33, 33
908 DATA 32, 35, 31, 34, 33
909 DATA 41, 34, 40, 41, 42
910 DATA 41, 40, 43, 37, 41
911 DATA 37, 42, 36, 41, 37
912 DATA 45, 38, 31, 36, 43
10000 END
```

Sample Run of SIP300

```
 BASIC SIP300
RUN
WHAT IS NUMBER OF LOCALITIES?
!3
WHAT IS NUMBER OF SUBSAMPLES PER LOCALITY?
! 4
WHAT IS NUMBER OF SPECIMENS IN EACH SUBSAMPLE?
!5
               SUM ØF           DEG. OF          MEAN
SØURCE         SQUARES          FREEDOM          SQUARE

LOCALITIES      1084.3             2           542.148

SAMPLES           35.75            9             3.97222

SPECIMENS        330.797          48             6.8916

TOTALS          1450.84           59


F-TESTS:
  FOR LOCALITIES, F=   136.485
    WITH DEGREES ØF FREEDOM:     2 AND     9

  FOR SAMPLES, F=      .576386
    WITH DEGREES ØF FREEDØM:     9 AND    48
```

SIP306

```
001 REM  SIP306 CALCULATES CHI SQUARE TO TEST GOODNESS-OF-FIT FOR
002 REM  FREQUENCY DATA.  THE ASSUMPTION IS MADE THAT THEORETICAL
004 REM  FREQUENCIES HAVE BEEN CALCULATED AS SHOWN IN EXAMPLE ON
005 REM  P. 308.  IN DATA LINE 900 AND FOLLOWING, PUT THE FIRST OBSERVED
006 REM  OR FIRST LUMPED FREQUENCIES, THEN THE FIRST EXPECTED FREQUENCY,
007 REM  THEN THE SECOND "O", THE SECOND "E", AND SO ON (DATA FROM
008 REM  EXAMPLE ON P. 308).  TERMINATE WITH VALUE 5.0E25.
```

```
009 REM
020 LET B = 0
025 LET C = 0
040 READ Ø, E
050 IF Ø=5.0E25 THEN 300
060 LET A=((Ø-E)↑2)/E
090 LET B=B+A
100 LET C=C+1
110 GØ TØ 40
300 PRINT "CHI SQUARE=";B
305 PRINT
310 PRINT "DEGREES OF FREEDOM=";C-3
900  DATA 4, 4.47, 11, 9.87, 18, 18.37, 21, 21.90, 20, 17.59, 9
901 DATA 9.48, 3, 4.05, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP306

```
CHI SQUARE=        .849948

DEGREES OF FREEDOM=      4
```

SIP309

```
001 REM  SIP309 USES THE DATA FROM SIP129 TO CALCULATE CHI SQUARE FOR
002 REM  A POISSON DISTRIBUTION, AND MUST BE MERGED WITH IT TO RUN.
003 REM  THIS PROGRAM CANNOT BE RUN INDEPENDENTLY.  TEST DATA IN PRO-
004 REM  GRAM TAKEN FROM P. 310, EXAMPLE 86.
363 LET N=0
365 PRINT"FOR AN","     POISSON","     THEORETICAL"
370 PRINT"X VALUE","PROBABILITY","     FREQUENCY"
375 PRINT"  OF:","          IS:","        IS:"
390 IF X=5.0E25 THEN 590
395 LET N=N+1
520 PRINT F, J, J*C
540 LET S=J*C
550 PRINT F, E, E*C
560 LET S=E*C
565 GØ TØ 580
570 PRINT F, D*E, D*E*C
575 LET S=D*E*C
580 LET T=T+(((Y-S)↑2)/S)
585 GØ TØ 380
590 PRINT
595 PRINT "CHI SQUARE=";T
600 PRINT
610 PRINT "DEGREES OF FREEDOM=";N-2
```

Sample Run of SIP309

```
      MEAN=          .733334
VARIANCE=          1.02989

FOR AN               POISSON          THEORETICAL
X VALUE          PROBABILITY          FREQUENCY
  OF:                     IS              IS:
       0               .480306        14.4092

       1               .352224        10.5667

       2               .129149         3.87447

       3           .315697E-01          .947092

       4           .578778E-02          .173633

       5           .848874E-03     .254662E-01

CHI SQUARE=         4.56661

DEGREES OF FREEDOM=         4
```

SIP312

```
001 REM  SIP312 RUNS THE VARIANCE RATIO TEST FOR EITHER A POISSON OR
002 REM  A NORMAL DISTRIBUTION. DATA IN PROGRAM IS FOR A P
003 REM  POISSON DISTRIBUTION.  DATA SHOULD GO IN LINE 900 AS
004 REM  FOLLOWS: VALUE OF FIRST CATEGORY, THEN OBSERVED FREQUENCY
005 REM  OF THAT CATEGORY, THEN SECOND CATEGORY AND FREQUENCY,
006 REM  FINALLY TERMINATED WITH 5.0E25 REPEATED ONCE.
007 REM
080 PRINT "TYPE 1 FOR POISSON DISTRIBUTION, 2 FOR NORMAL DIST."
085 INPUT Z
100 READ X,Y
110 IF Y=0 THEN 100
130 IF X=5.0E25 THEN 180
140 LET B=B+(X*Y)
160 LET C=C+Y
165 LET W=W+(Y*X*X)
170 GØ TØ 100
180 LET D=B/C
190 LET E=((W)-(C*D*D))/(C-1)
195 PRINT
196 PRINT
197 IF Z=2 THEN 215
200 LET F=(E*C)/D
210 LET G=C-1
211 GØ TØ 220
215 LET F=(E*C)/((D*(C-D))/C)
```

```
220 PRINT "SAMPLE MEAN=";D
225 PRINT "OBSERVED VARIANCE=";E
226 PRINT
230 PRINT "CHI SQUARE IS";F;"  WITH";G;"DEGREES OF FREEDOM"
900 DATA 0, 16, 1, 9, 2, 3, 3, 1, 4, 1, 5.0E25
999 DATA 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP312

```
TYPE 1 FOR POISSON DISTRIBUTION, 2 FOR NORMAL DIST.
!1


SAMPLE MEAN=        .733334
OBSERVED VARIANCE=       1.02989

CHI SQUARE IS     42.1317  WITH     29 DEGREES OF FREEDOM
```

SIP313–SIP338

The tests shown in these pages, including the small sample correction (p. 322) and the list of expected values are in program SIP318. Data organized as on page 330 must be run in program SIP331.

SIP318

```
001 REM  SIP318 WILL HANDLE ALL SIZES OF CONTIGENCY TABLES. DATA GOES
002 REM  IN LINE 900.  START WITH FIRST ROW, PUTTING IN ALL OBSERVED
003 REM  VALUES ACROSS TABLE, FOLLOWED IMMEDIATELY BY ALL OBSERVED
004 REM  VALUES IN SECOND ROW, AND SO ON.  BE PREPARED TO INDICATE
005 REM  FIRST THE NUMBER OF ROWS, THEN THE NUMBER OF COLUMNS.
006 REM  DATA IN PROGRAM TAKEN FROM EXAMPLE 91, P. 322.
007 REM
100 PRINT "WHAT IS NUMBER OF ROWS?"
110 INPUT A
120 PRINT "WHAT IS NUMBER OF COLUMNS?"
130 INPUT B
140 FØR I=1 TØ A
150 FØR J=1 TØ B
160 READ Z(I,J)
170 LET C=C+Z(I,J)
180 NEXT J
190 LET V(I)=C
200 LET E=E+C
210 LET C=0
220 NEXT I
230 FØR J=1 TØ B
240 FØR I=1 TØ A
250 LET D=D+Z(I,J)
260 NEXT I
```

```
280 LET U(J)=D
290 LET D=0
300 NEXT J
305 PRINT "IF YOU WISH TO USE SMALL SAMPLE CØRRECTION (P. 322)"
306 PRINT "  TYPE 2, IF NOT, TYPE 1"
307 INPUT Q
310 FØR I=1 TØ A
330 FØR J=1 TØ B
340 LET W(I,J)=(V(I)*U(J))/E
350 LET T=Z(I,J)*Z(I,J)/W(I,J)
360 LET K=K+T
370 NEXT J
380 NEXT I
385 LET K=K-E
390 LET F=(A-1)*(B-1)
395 IF Q=2 THEN515
400 PRINT "CHI SQUARE IS";K;"WITH";F;"DEGREES OF FREEDOM"
410 PRINT
420 PRINT"IF YOU WANT A LIST OF THE CALCULATED EXPECTED VALUES"
421 PRINT"  TYPE 2, IF NOT, TYPE 1"
430 INPUT Q
440 IF Q=2 GØ TØ 460
450 GØ TØ 10000
460 FØR I=1 TØ A
470 FØR J=1 TØ B
480 PRINT I;"RØW,";J;" CØLUMN=";W(I,J)
490 NEXT J
500 NEXT I
510 RETURN
515 IF (Z(1,1)*Z(2,2))<(Z(2,1)*Z(1,2)) THEN 525
520 LET P=(Z(1,1)*Z(2,2))-(Z(2,1)*Z(1,2))-(E/2)
522 GØ TØ 530
525 LET P=(Z(2,1)*Z(1,2))-(Z(1,1)*Z(2,2))-(E/2)
530 LET P=(E*(P*P))/(U(1)*U(2)*V(1)*V(2))
540 PRINT "CHI SQUARE IS:";P;"WITH  1 DEGREE OF FREEDOM"
550 GØ TØ 410
900 DATA 1,10,8,3
999 DATA 5.0E25
1000 END
```

Sample Run of SIP318

```
WHAT IS NUMBER OF ROWS?
!2
WHAT IS NUMBER OF COLUMNS?
!2
IF YOU WISH TO USE SMALL SAMPLE CØRRECTION (P. 322)
  TYPE 2, IF NOT, TYPE 1
!2
CHI SQUARE IS:     6.76923WITH  1 DEGREE OF FREEDOM
```

```
IF YOU WANT A LIST OF THE CALCULATED EXPECTED VALUES
  TYPE 2, IF NOT, TYPE 1
!2
     1 ROW,      1  COLUMN=      4.5
     1 ROW,      2  COLUMN=      6.5
     2 ROW,      1  COLUMN=      4.5
     2 ROW,      2  COLUMN=      6.5
```

SIP331

```
001 REM  SIP331 COMBINES A SERIES OF KNOWN CHI SQUARES TO CALCULATE
002 REM  THE STANDARDIZED NORMAL DEVIATE.  THE DATA, ENTERED IN LINE
003 REM  900, ARE IN THIS ORDER: FOR FIRST SAMPLE, FIRST PERCENTAGE,
004 REM  SECOND PERCENTAGE, AND CHI SQUARE FOR FIRST SAMPLE (FROM
005 REM  SIP318 FOR EACH 2X2 TEST9, THEN SAME VALUES FOR SECOND SAMPLE,
006 REM  ETC.  TERMINATE WITH 5.0E25 AND TWO ZEROES.  DATA IN PROGRAM
007 REM  FROM EXAMPLE 95, P. 330.
100 LET C=0
110 READ  X, Y, Z
120 IF X=5.0E25 THEN 200
130 LET C=C+1
140 IF Y>X THEN 170
150 LET A=SQR(Z)
160 GØ TØ 180
170 LET A=SQR(Z)
175 LET A=-A
180 LET B=B+A
190 GØ TØ 110
200 PRINT
210 PRINT "STANDARDIZED NORMAL DEVIATE="; B/SQR(C)
900 DATA 30.6, 36.7, 1.18, 52.9, 53.4, 1.20, 42.9, 41.9, .18, 39.6
901 DATA 34.4, .69, 50.0, 44.3, .62, 53.1, 48.9, 12.06, 5.0E25, 0, 0
999 DATA 5.0E25, 5.0E25
1000 END
```

Sample Run of SIP331

```
STANDARDIZED NORMAL DEVIATE=      1.36064
```

SIP339–SIP372

Most of the material in this chapter is not amenable to time-shared computer use, although plotters that can be attached to computers, including time-shared machines, are available. No attempt is made here to program for plotting. An X,Y plot program usually is standard in library programs of time-share companies. One operation discussed in this chapter is easily performed by the computer, however, and advantage should be taken of it in graphs using log or semilog plots. SIP357 will convert all given values into logs, either natural logs or logs to the base 10.

SIP357

```
001 REM  SIP357 CONVERTS ACTUAL VALUES TO LOG VALUES.  YOU HAVE A
002 REM  CHOICE OF NATURAL LOGS OR LOGS TO A BASE 10.  PUT YOUR DATA
003 REM  IN LINE 900, TERMINATING WITH THE VALUE 5.0E25. DATA FROM
004 REM  STANDARD LENGTHS, EXAMPLE 102, P. 399, WITH LOGS AS LISTED
005 REM  IN EXAMPLE 103, P. 403.
006 REM
080 PRINT
090 PRINT
100 PRINT "FOR NATURAL LOGS, TYPE 1, FOR BASE 10 LOGS TYPE 2"
110 INPUT A
120 PRINT
130 PRINT "ACTUAL VALUE","    LOG VALUE"
135 PRINT
140 IF A=2 THEN 200
150 READ X
160 IF X=5.0E25 THEN 1000
170 PRINT X, LØG(X)
180 GØ TØ 150
200 READ X
210 IF X=5.0E25 THEN 1000
220 PRINT X,(.4342945*LØG(X))
230 GØ TØ 200
900 DATA 37.6, 44.8, 54.1, 64.5, 74.1, 84.0, 93.1, 106, 116
901 DATA 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP357

```
FOR NATURAL LOGS, TYPE 1, FOR BASE 10 LOGS TYPE 2
!2

ACTUAL VALUE      LOG VALUE

    37.6            1.57519
    44.8            1.65128
    54.1            1.7332
    64.5            1.80956
    74.1            1.86982
    84              1.92428
    93.1            1.96895
   106              2.02531
   116              2.06446
```

SIP391

```
001 REM  SIP391 WILL CALCULATE BOTH THE REGRESSION COEFFICIENT OF
002 REM  Y ('K') AND THE GEOMETRIC RATE OF INCREASE ('K SUB G')
```

```
003 REM  FOR A GROWTH CURVE.  DATA GOES IN LINE 900, WITH TIME (7T')
004 REM  VALUE FIRST AND MEASUREMENT ('Y') VALUE SECOND, CONTINUING
005 REM  THROUGH SERIES.  YOU WILL BE ASKED FOR THE NUMBER OF ROWS.
006 REM  EXAMPLE IN PROGRAM FROM P. 392.
090 PRINT
100 PRINT "WHAT IS NUMBER OF ROWS?"
110 INPUT A
112 PRINT
115 PRINT "     T","     Y","INCREMENT","     K","     K-G"
120 FØR I=0 TØ (A-1)
130 READ T(I), Y(I)
135 IF I<>0 THEN 140
138 PRINT T(I), Y(I)
140 NEXT I
150 FØR I=1 TØ (A-1)
160 LET K(I)=(Y(I)-Y(I-1))/(T(I)-T(I-1))
170 LET G(I)=(LØG(Y(I))-LØG(Y(I-1)))/(T(I)-T(I-1))
180 PRINT T(I),Y(I),Y(I)-Y(I-1),K(I),G(I)
190 NEXT I
900 DATA 0, 36.1, 14, 53.4, 28, 68.1, 43, 79.3, 57, 87.3
1000 END
```

Sample Run of SIP391

```
WHAT IS NUMBER OF ROWS?
!5

   T              Y          INCREMENT          K              K-G
  0             36.1
  14             53.4           17.3          1.23571      .279655E-01
  28             68.1           14.7          1.05         .173690E-01
  43             79.3           11.2           .746667     .101508E-01
  57             87.3            8             .571429     .686516E-02
```

SIP410

```
001 REM  SIP410 CALCULATES THE VALUE OF THE COEFFICIENT OF ALLOMETRY
002 REM  ("ALPHA") FOR GROWTH PERIODS, USING THE GROWTH EQUATION.
003 REM  PUT DATA IN LINE 900, WITH VALUES IN X-Y PAIRS (NOTE THAT
004 REM  THESE DATA CAN BE PUT IN SIP233 IF THE "BEST FIT" CURVE IS
005 REM  NEEDED).  PRØGRAM WILL NUMBER PAIRS STARTING WITH ONE, AND
006 REM  PRINT RESULTS FOR EACH PERIOD ('T') CHANGE.  PROGRAM WILL
007 REM  CALL FOR TOTAL NUMBER OF PERIODS.  DATA IN PROGRAM FROM
008 REM  EXAMPLE 105, P. 410.
009 REM
090 PRINT
100 PRINT "WHAT IS NUMBER OF GROWTH PERIODS?"
110 INPUT Z
115 PRINT
```

```
120 PRINT "       GROWTH PERIOD","        ALPHA"
130 PRINT
140 FØR I=1 TØ Z
150 READ X(I), Y(I)
160 IF X(I)=5.0E25 THEN 1000
170 IF I=1 THEN 250
180 LET A=(LØG(Y(I))-LØG(Y(I-1)))/(LØG(X(I))-LØG(X(I-1)))
200 PRINT (I-1)"--"I, A
250 NEXT I
900 DATA 2.138, .818, 3.188, .747, 4.792, .683, 6.245, .681
901 DATA 8.648, .572, 11.561, .447, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP410

```
WHAT IS NUMBER OF GROWTH PERIODS?
!6

      GROWTH PERIOD                 ALPHA

     1         --       2            -.227264
     2         --       3            -.219775
     3         --       4        -.110730E-01
     4         --       5            -.535788
     5         --       6            -.84937
```

SIP411

```
001 REM  SIP411 RUNS THE RICHARDS AND KAVANAUGH TEST FOR DETECTING
002 REM  DEVIATIONS FROM SIMPLE ALLOMETRY.  DATA FOR A PAIR OF
003 REM  COUNTS OR MEASUREMENTS GOES IN LINE 900, WITH X AND Y
004 REM  VALUES FOR EACH INDIVIDUAL IN SAMPLE.  END DATA STRING WITH
005 REM  VALUE 5.0E25.  DATA IN PROGRAM FROM EXAMPLE 106, P. 412.
006 REM
100 PRINT "      LOG X",  "        LOG Y","  CALC. LOG Y","  DEVIATION"
110 PRINT
120 READ X, Y
130 IF X=5.0E25 THEN 1000
140 PRINT LØG(X), LØG(Y), (-1.00+.70*LØG(X)),(LØG(Y)-(-1.00+.70*LØG(X)))
150 GØ TØ 120
900 DATA .9, .3, 1.3, .4, 1.1, .3, 1.6, .5, 1.7, .6, 1.9, .6, 2.1, .6
901 DATA 2.1, .7, 2.6, .8, 2.6, .8, 6.6, 1.3, 4.3, 1.1, 4.3, 1.3, 6
902 DATA 1.3, 6, 1.3, 8, 1.6, 11.1, 1.7, 11.9, 1.7, 5.0E25, 5.0E25
999 DATA 5.0E25
1000 END
```

Sample Run of SIP411

LOG X	LOG Y	CALC. LOG Y	DEVIATION
-.105361	-1.20397	-1.07375	-.13022
.262364	-.916291	-.816346	-.999456E-01
.953101E-01	-1.20397	-.933283	-.27069
.470003	-.693147	-.670998	-.221493E-01
.530628	-.510826	-.628561	.117735
.641853	-.510826	-.550703	.398770E-01
.741937	-.510826	-.480644	-.301814E-01
.741937	-.356675	-.480644	.123969
.955511	-.223144	-.331142	.107999
.955511	-.223144	-.331142	.107999
1.88707	.262364	.320948	-.585844E-01
1.45861	.953101E-01	.210300E-01	.742801E-01
1.45861	.262364	.210300E-01	.241334
1.79176	.262364	.254231	.813293E-02
1.79176	.262364	.254231	.813293E-02
2.07944	.470003	.455609	.143944E-01
2.40694	.530628	.684861	-.154233
2.47654	.530628	.733577	-.202949

Index

Publication in *Smithsonian Contributions to Zoology*

Manuscripts for serial publications are accepted by the Smithsonian Institution Press, subject to substantive review, only through departments of the various Smithsonian museums. Non-Smithsonian authors should address inquiries to the appropriate department. If submission is invited, the following format requirements of the Press will govern the preparation of copy. (An instruction sheet for the preparation of illustrations is available from the Press on request.)

Copy must be typewritten, double-spaced, on one side of standard white bond paper, with 1½" top and left margins, submitted in ribbon copy with a carbon or duplicate, and accompanied by the original artwork. Duplicate copies of all material, including illustrations, should be retained by the author. There may be several paragraphs to a page, but each page should begin with a new paragraph. Number consecutively all pages, including title page, abstract, text, literature cited, legends, and tables. The minimum length is 30 pages of typescript and illustrations.

The *title* should be complete and clear for easy indexing by abstracting services. Taxonomic titles will carry a final line indicating the higher categories to which the taxon is referable: "(Hymenoptera: Sphecidae)." Include an *abstract* as an introductory part of the text. Identify the *author* on the first page of text with an unnumbered footnote that includes his professional mailing address. A *table of contents* is optional. An *index,* if required, may be supplied by the author when he returns page proof.

Two *headings* are used: (1) text heads (boldface in print) for major sections and chapters and (2) paragraph sideheads (caps and small caps in print) for subdivisions. Further headings may be worked out with the editor.

In *taxonomic keys,* number only the first item of each couplet; if there is only one couplet, omit the number. For easy reference, number also the taxa and their corresponding headings throughout the text; do not incorporate page references in the key.

In *synonymy,* use the short form (taxon, author, date, page) with a full reference at the end of the paper under "Literature Cited." Begin each taxon at the left margin with subsequent lines indented about three spaces. Within a taxon, use a period-dash (.—) to separate each reference. Enclose with square brackets any annotation in or at the end of the taxon. For *references within the text,* use the author-date system: "(Jones, 1910)" or "Jones (1910)." If the reference is expanded, abbreviate the data: "Jones (1910, p. 122, pl. 20: fig. 1)."

Simple *tabulations* in the text (e.g., columns of data) may carry headings or not, but they should not contain rules. Formal *tables* must be submitted as pages separate from the text, and each table, no matter how large, should be pasted up as a single sheet of copy.

For *measurements and weights,* use the metric system instead of (or in addition to) the English system.

Illustrations (line drawings, maps, photographs, shaded drawings) can be intermixed throughout the printed text. They will be termed *Figures* and should be numbered consecutively; however, if a group of figures is treated as a single figure, the individual components should be indicated by lowercase italic letters on the illustration, in the legend, and in text references: "Figure 9*b*." If illustrations (usually tone photographs) are printed separately from the text as full pages on a different stock of paper, they will be termed *Plates,* and individual components should be lettered (Plate 9*b*) but may be numbered (Plate 9: figure 2). Never combine the numbering system of text illustrations with that of plate illustrations. Submit all legends on pages separate from the text and not attached to the artwork.

In the *bibliography* (usually called "Literature Cited"), spell out book, journal, and article titles, using initial caps with all words except minor terms such as "and, of, the." (For capitalization of titles in foreign languages, follow the national practice of each language.) Underscore (for italics) book and journal titles. Use the colon-parentheses system for volume, number, and page citations: "10(2):5–9." Spell out such words as "figures" and "plates" (or "pages" when used alone).

For *free copies* of his own paper, a Smithsonian author should indicate his requirements on "Form 36" (submitted to the Press with the manuscript). A non-Smithsonian author will receive 50 free copies; order forms for quantities above this amount with instructions for payment will be supplied when page proof is forwarded.

SMITHSONIAN INSTITUTION
WASHINGTON, D.C.